

JASECI

BIBLE

Jason Mars, ~~PhD~~ Ninja

v1.3

I welcome you, neophyte, to embark on the journey of becoming a true Jaseci Ninja!

Btw, this is a silly book, please take that seriously :-)

Contents

Preface	8
1 Introduction	9
I World of Jaseci	10
2 What and Why is Jaseci?	11
2.1 Viewing the Problem Landscape Spacially	11
2.2 Compute via The Collective, Ants in the Colony	11
3 Abstrations of Jaseci	12
3.1 Graphs, the Friend that Never Gets Invited to the Party	12
3.1.1 Yes, But What Kind of Graphs	13
3.1.2 Putting it All Into Context	15
3.2 Walkers	15
3.3 Abilities	15
3.4 Other Abstractions Not Yet Actualized in Current Jaseci	15
4 Architecture of Jaseci and Jac	16
4.1 Anatomy of a Jaseci Application	16
4.2 The Jaseci Machine	16
4.2.1 Machine Core	16
4.2.2 Jaseci Cloud Server	16
5 Interfacing a Jaseci Machine	17
5.1 JSCTL: The Jaseci Command Line Interface	18
5.1.1 The Very Basics: CLI vs Shell-mode, and Session Files	19
5.1.2 A Simple Workflow for Tinkering	22
5.2 Jaseci REST API	27
5.3 Full Spec of Jaseci Core APIs	27
5.3.1 Cheatsheet	27

II	The Jac Programming Language	31
6	Jac Language Overview and Basics	32
6.1	The Obligatory Hello World	33
6.2	Numbers, Arithmetic, and Logic	34
6.2.1	Basic Arithmetic Operations	34
6.2.2	Comparison, Logical, and Membership Operations	35
6.2.3	Assignment Operations	37
6.2.4	Precedence	38
6.2.5	Primitive Types	38
6.3	Foreshadowing Unique Graph Operations	40
6.4	More on Strings, Lists, and Dictionaries	42
6.4.1	Library of String Operations	44
6.4.2	Library of List Operations	44
6.4.3	Library of Dictionary Operations	44
6.5	Control Flow	44
7	Graphs, Architypes, and Walkers in Jac	48
7.1	Structure of a Jac Program	48
7.2	Graphs as First Class Citizens	49
7.2.1	Connect and Spawn operations	49
7.2.2	Static Graph Creation	51
7.3	Walkers as the second First Class Citizens	55
7.4	Architypes	57
7.4.1	Context on Nodes and Edges	57
7.4.2	Copy Assignment Operator	59
7.4.3	Plucking Values from Node and Edge Sets	60
7.4.4	Referencing and Dereferencing Nodes and Edges	61
7.5	Actions and Abilities	62
7.5.1	Actions	62
7.5.2	Fused Interactions Between Nodes and Actions	63
7.5.3	Abilities	64
7.5.4	here and visitor, the ‘this’ references of Jac	66
7.6	Inheritance	66
8	Walkers Navigating Graphs	67
8.1	Taking Edges (and Nodes?)	67
8.1.1	Basic Walks	67
8.1.2	Breadth First vs Depth First Walks	69
8.2	Ignoring and Deleting	70
8.3	Reporting Back as you Travel	71
8.4	Yielding Walkers	72
8.4.1	Yield Shorthands	72
8.4.2	Technical Semantics of Yield	73

8.4.3 Walkers Yielding Other Walkers (i.e., Yielding Deeply)	73
9 Actions and Action Sets	75
9.1 Standard Action Library	75
9.1.1 date	75
9.1.2 file	75
9.2 Building Your Own Library	76
10 Imports, File I/O, Tests, and More	77
10.1 Tests in Jac	77
10.2 Imports	78
10.3 File I/O	79
10.4 Visualizing Graph with Dot Output	80
 III Jaseci AI Kit	 81
 IV Crafting Jaseci	 82
11 Architecting Jaseci Core	83
12 Architecting Jaseci Cloud Serving	84
Epilogue	85
 A Jumping Right In, TLDR sytle	 86
A.1 Installation	86
A.1.1 Python Environment	87
A.1.2 Installing Jaseci	88
A.1.3 VSCode and the Jac Language Extension	90
A.2 Coding in Jac	91
A.2.1 Jac Basics	91
A.2.2 Types in Jac	93
A.2.3 Fun with Lists and Dictionaries	93
A.2.4 Control Flow	94
A.2.5 Graphs in Jac	95
A.2.6 Navigating Graphs with Walkers	96
A.2.7 Compute in Nodes	97
A.2.8 Static Graphs	99
A.2.9 Writing Tests	100
A.3 Jac Hacking Workflow	101
A.3.1 Using Imports	102
A.3.2 Leveraging Static Graphs for Quick Prototyping	103
A.3.3 Test Driven Development	104

A.3.4	File I/O	104
A.3.5	Building to JIR	105
A.4	AI with Jaseci Kit	105
A.4.1	Installing Jaseci Kit	105
A.4.2	Loading Actions from Jaseci Kit	106
A.4.3	Using AI in Jac	107
A.5	Launching a Jaseci Web Server	108
A.6	Deploying Jaseci at Scale	108
A.6.1	Quick-start with Kubectl	108
A.6.2	Managing Jac in Cloud	108
B	Rants	109
B.1	Utilizing Whitespace for Scoping is Criminal (Yea, I'm looking at you Python)	109
B.2	“Using ‘master’ in Git is NOT Racist!” by a black dude	109
C	Full Jac Grammar Specification	110

Preface

The way we design and write software to do computation and AI today is poop. How poopy you ask? Hrm... let me think... In my approximation, if you were to use it as a fuel source, it would be able to run all the blockchain transactions across the aggregate of current and future coins for a decade.

Hrm, too much? Probably. I guess you'd expect me to use concrete examples and cite evidence to make my points. I mean, I could write something like "*The imperative programming model utilized in near all of the production software produced in the last four decades has not fundamentally changed since blah blah blah...*". I'd certainly sound more credible and such. Well, though I have indeed grown accustomed to writing that way, boy has it gotten old.

I'm not going to do that for this book. Let's have fun. After all, Jaseci has never been work for me, its play (and art). Very ambitious play granted, but play at it's core.

Everything here is based on my opinion...no, expert *ninja* opinion, and my intuition. That suffices for me, and I hope it does for you. Even though I have spent decades coding and leading teams of coders working on the holy grail technical challenges of our time, I won't rely on that to assert credibility. Lets let these ideas stand or die on their own merit. Its my gut that tells me that we can do better. This book describes my attempt at better. I hope you find value in it. If you do, awesome! If you don't, awesome!

Chapter 1

Introduction

Coming soon!

Part I

World of Jaseci

Chapter 2

What and Why is Jaseci?

2.1 Viewing the Problem Landscape Spacially

2.2 Compute via The Collective, Ants in the Colony

Chapter 3


Abstractions of Jaseci

3.1 Graphs, the Friend that Never Gets Invited to the Party


There's something quite strange that has happened with our common languages over the years, ...decades. When you look at it, almost every data structure we programmers use to solve problems can be modeled formally as a graph, or a special case of a graph, (save perhaps hash tables). Think about it, stacks, lists, queues, trees, heaps, and yes, even graphs, can be modeled with graphs. But, low and behold, no common language utilizes the formal semantics of a graph as its first order abstraction for data or memory. I mean, isn't it a bit odd that practically every data structure covered in the language-agnostic classic foundational work *Introduction to Algorithms* [4] can most naturally be reasoned about as a graph, yet none of the common languages have built in and be designed around this primitive. I submit that the graph semantic is stupidly rich, very nice for us humans to reason about, and, most importantly for the purpose of Jaseci, is inherently well suited for the conceptualization and reasoning about computational problems, especially AI problems.

There are a few arguments that may pop into mind at this point of my conjecture.

- “Well there are graph libraries in my favorite language that implement graph semantics, why would I need a language to force the concept upon me?” or
- “Duh! Interacting with all data and memory through graphical abstractions will make the language sssloooowww as hell since memory in hardware is essentially a big array, what is this dude talking about!?!?”



(a) Directed graph with cycle between nodes three and four.



(b) Multigraph with parallel edges and self-loops

Figure 3.1: Examples of first order graph symantics supported by Jaseci.¹

For the former of these two challenges, I counter with two points. First, the core design languages are always based upon their inherent abstractions. With graphs not being one such abstraction, the language’s design will not be optimized to empower programmers to nimbly do gymnastics with the rich language symantics that correspond to the rich semantics graphs offer (You’ll see what I mean in later chapters).

For the latter question, I’d respond, “Have you SEEN the kind of abstractions in modern languages?!? It’s ridiculous, lets look at python dictionaries, actually scratch that, lets keep it simple and look at dynamic typing in general. The runtime complexity to support dynamic typing is most certainly hgiher than what would be needed to support graph symantics. Duh right back at’ya!”

3.1.1 Yes, But What Kind of Graphs

There are many categories of graphs to consider when thinking about the abstractions to support in Jaseci. There are rules to be defined as to the availabe semantics of the graphs. Should all graphs be directed graphs, should we allow the creation of undirected graphs, what about parallel edges or multigraph, are those explicitly expressible or discouraged / banned, can we express hypergraph, and what combination of these graphical sematics should be able to be manifested and manipulated through the programming model. At this point I can feel your eyes getting droopy and your mind moving into that intermediary state between concious and sleeping, so let me cut to the answer.

¹Images credits to wiki contributors [2, 3]

In Jaseci, we elect to assume the following semantics:

1. Graphs are directed (as per Figure 3.1a) with a special case of a doubly directed edge type which can be utilized practically as an undirected edge (imagine fusing the two edges between nodes 3 and 4 in the figure).
2. Both nodes and edges have their own distinct identities (i.e. an edge isn't representable as a pairing of two nodes). This point is important as both nodes and edges can have contexts.
3. Multigraphs (i.e., parallel edges) are allowed, including self-loop edges (as per Figure 3.1b).
4. Graphs are not required to be acyclic.
5. No hypergraphs, as I wouldn't want Jaseci programmers heads to explode.

As an aside, I would describe Jaseci graphs as strictly unstrict directed multigraphs that leverages the semantics of parallel edges to create a laymans 'undirected edge' by shorthand-ing two directed edges pointed in opposite directions between the same two nodes.

Nerd Alert 1 (*time to let your eyes glaze over*)

I'd formally describe a Jaseci Graph as an 7-tuple $(N, E, C, s, t, c_N, c_E)$, where

1. N is the set of nodes in a graph
2. E is the set of edges in a graph
3. C is the set of all contexts
4. $s: E \rightarrow V$, maps the source node to an edge
5. $t: E \rightarrow V$, maps the target node to an edge
6. $c_N: N \rightarrow C$, maps nodes to contexts
7. $c_E: E \rightarrow C$, maps edges to contexts

An undirected edge can then be formed with a pair of edges (x, y) if three conditions are met,

1. $x, y \in E$
2. $s(x) = t(y)$, and $s(y) = t(x)$
3. $c_E(x) = c_E(y)$

If you happen to have read that formal definition and didn't enter deep comatose you may be wondering "Whoa, what was that context stuff that came outta nowhere! What's this guy trying to do here, sneaking a new concept in as if it was already introduced and described."

Worry not friend, lets discuss.

3.1.2 Putting it All Into Context

A key principle of Jaseci is to reshape and reimagine how we view data and memory. We do so by fusing the concept of data with the intuitive and rich semantics of graphs as the lowest level primitive to view memory.

Nerd Alert 2 (*time to let your eyes glaze over*)

A context is a representation of data that can be expressed simply as a 3-tuple (\sum_K, \sum_V, p_K) , where

1. \sum_K is a finite alphabet of keys
2. \sum_V is a finite alphabet of values
3. p_K is the pairing of keys to values

3.2 Walkers

3.3 Abilities

3.4 Other Abstractions Not Yet Actualized in Current Jaseci

Chapter 4

Architecture of Jaseci and Jac

4.1 Anatomy of a Jaseci Application

4.2 The Jaseci Machine

4.2.1 Machine Core

4.2.2 Jaseci Cloud Server

Chapter 5

Interfacing a Jaseci Machine

Now that we know what Jaseci is all about, next lets roll up our sleeves and jump in. One of the best ways to jump into Jaseci world is to gather some sample Jac programs and start tinkering with them.

Before we jump right into it, it's important to have a bit of an understanding of the the way the interface itself is architected from in the implementation of the Jaseci stack. Jaseci has a module that serves as its the core interface (summarized in Table 5.1) to the Jaseci machine. This interface is expressed as a set of method functions within

a python class in Jaseci called `master`. (By the way, don't worry, it's ok to use "master", its not racist, see Rant B.2 for more context). The 'client' expressions of that interface in the forms of a command line tool `jsctl` and a server-side REST API built using Django ¹. Figure 5.1 illustrates this architecture representing the relationship between core APIs and client side expressions.




Figure 5.1: Jaseci Interface Architecture

If I may say so myself the code architecture of interface generation from function signatures is elegant, sexy, and takes advantage of the best python has to offer in terms of its support for introspection. With this approach, as the set of functions and their semantics change in the `master` API class, both the JSCTL Cli tool and the REST Server-side API changes. We dig into this and tons more in the Part IV, so we'll leave the discussion on implementation

¹Django [5] is a Python web framework for rapid development and clean, pragmatic design

architecture there for the moment. Lets jump right into how we get started playing with some leet Jaseci haxoring. First we start with JSCTL then dive into the REST API.

5.1 JSCTL: The Jaseci Command Line Interface

JSCTL or `jsctl` is a command line tool that provides full access to Jaseci. This tool is installed alongside the installation of the Jaseci Core package and should be accessible from the command line from anywhere. Let's say you've just checked out the Jaseci repo and you're in head folder. You should be able to execute the following.

```
haxor@linux:~/jaseci# pip3 install ./jaseci_core
Processing ./jaseci_core
...
Successfully installed jaseci-0.1.0
haxor@linux:~/jaseci# jsctl --help
Usage: jsctl [OPTIONS] COMMAND [ARGS]...

    The Jaseci Command Line Interface

Options:
  -f, --filename TEXT Specify filename for session state.
  -m, --mem-only Set true to not save file for session.
  --help Show this message and exit.

Commands:
  alias Group of 'alias' commands
  archetype Group of 'archetype' commands
  check Group of 'check' commands
  config Group of 'config' commands
  dev Internal dev operations
  edit Edit a file
  graph Group of 'graph' commands
  login Command to log into live Jaseci server
  ls List relevant files
  object Group of 'object' commands
  sentinel Group of 'sentinel' commands
  walker Group of 'walker' commands
haxor@linux:~/jaseci#
```

Here we've installed the Jaseci python package that can be imported into any python project with a directive such as `import jaseci`, and at the same time, we've installed the `jsctl` command line tool into our OS environment. At this point we can issue a call to say `jsctl --help` for any working directory.

Nerd Alert 3 *(time to let your eyes glaze over)*

Python Code 5.1 shows the implementation of `setup.py` that is responsible for deploying the `jsctl` tool upon `pip3` installation of Jaseci Core.

Python Code 5.1: `setup.py` for Jaseci Core

```

1  from setuptools import setup, find_packages
2
3  setup(
4      name="jaseci",
5      version="0.1.0",
6      packages=find_packages(include=["jaseci", "jaseci.*"]),
7      install_requires=[
8          "click>=7.1.0,<7.2.0",
9          "click-shell>=2.0,<3.0",
10         "numpy>=1.21.0,<1.22.0",
11         "antlr4-python3-runtime>=4.9.0,<4.10.0",
12         "requests",
13         "flake8",
14     ],
15     package_data={
16         "": ["*.ini"],
17     },
18     entry_points={"console_scripts": ["jsctl=jaseci.jsctl.jsctl:main"]},
19 )

```

5.1.1 The Very Basics: CLI vs Shell-mode, and Session Files

This command line tool provides full access to the Jaseci core APIs via the command line, or a shell mode. In shell mode, all of the same Jaseci API functionality is available within a single session. To invoke shell-mode, simply execute `jsctl` without any commands and `jsctl` will enter shell mode as per the example below.

```
haxor@linux:~/jaseci# jsctl
Starting Jaseci Shell...
jaseci > graph create
{
  "context": {},
  "anchor": null,
  "name": "root",
  "kind": "generic",
  "jid": "urn:uuid:ef1eb3e4-91c3-40ba-ae7b-14c496f5ced1",
  "j_timestamp": "2021-08-15T15:15:50.903960",
  "j_type": "graph"
}
jaseci > exit
haxor@linux:~/jaseci#
```

Here we launched `jsctl` directly into shell mode for a single session and we can issue various calls to the Jaseci API for that session. In this example we issue a single call to `graph create`, which creates a graph within the Jaseci session with a single root node, then exit the shell with `exit`.

The exact behavior can be achieved without ever entering the shell directly from the command line as shown below.

```
haxor@linux:~/jaseci# jsctl graph create
{
  "context": {},
  "anchor": null,
  "name": "root",
  "kind": "generic",
  "jid": "urn:uuid:91dd8c79-24e4-4a54-8d48-15bee52c340b",
  "j_timestamp": "2021-08-15T15:40:12.163954",
  "j_type": "graph"
}
haxor@linux:~/jaseci#
```

All such calls to Jaseci's API (summarized in Table 5.1) can be issued either through shell-mode and CLI mode.

Session Files At this point, it's important to understand how sessions work. In a nutshell, a session captures the complete state of a jaseci machine. This state includes the status of memory, graphs, walkers, configurations, etc. The complete state of a Jaseci machine can be captured in a `.session` file. Every time state changes for a given session via the `jsctl` tool the assigned session file is updated. If you've been following along so far, try this.

```

haxor@linux:~/jaseci# ls *.session
js.session
haxor@linux:~/jaseci# jsctl graph list
[
  {
    "context": {},
    "anchor": null,
    "name": "root",
    "kind": "generic",
    "jid": "urn:uuid:ef1eb3e4-91c3-40ba-ae7b-14c496f5ced1",
    "j_timestamp": "2021-08-15T15:55:15.030643",
    "j_type": "graph"
  },
  {
    "context": {},
    "anchor": null,
    "name": "root",
    "kind": "generic",
    "jid": "urn:uuid:91dd8c79-24e4-4a54-8d48-15bee52c340b",
    "j_timestamp": "2021-08-15T15:55:46.419701",
    "j_type": "graph"
  }
]
haxor@linux:~/jaseci#

```

Note from the first call to `ls` we have a session file that has been created call `js.session`. This is the default session file `jsctl` creates and utilizes when called either in cli mode or shell mode. After listing session files, notices the call to `graph list` which lists the root nodes of all graphs created within a Jaseci machine's state. Note `jsctl` lists two such graph root nodes. Indeed these nodes correspond to the ones we've just created when contrasting cli mode and shell mode above. Having these two graphs demonstrates that across both instantiations of `jsctl` the same session, `js.session`, is being used. Now try the following.

```

haxor@linux:~/jaseci# jsctl -f mynew.session graph list
[]
haxor@linux:~/jaseci# ls *.session
js.session mynew.session
haxor@linux:~/jaseci#

```

Here we see that we can use the `-f` or `--filename` flag to specify the session file to use. In this case we list the graphs of the session corresponding to `mynew.session` and see the JSON representation of an empty list of objects. We then list session files and see that one was created for `mynew.session`. If we were to now type `jsctl --filename js.session graph list`, we would see a list of the two graph objects that we created earlier.

In-memory mode Its important to note that there is also an in-memory mode that can be created buy using the `-m` or `--mem-only` flags. This flag is particularly useful when

you'd simply like to tinker around with a machine in shell-mode or you'd like to script some behavior to be executed in Jac and have no need to maintain machine state after completion. We will be using in memory session mode quite a bit, so you'll get a sense of its usage throughout this chapter. Next we actually see a workflow for tinkering.

5.1.2 A Simple Workflow for Tinkering

As you get to know Jaseci and Jac, you'll want to try things and tinker a bit. In this section, we'll get to know how `jsctl` can be used as the main platform for this play. A typical flow will involve jumping into shell-mode, writing some code, running that code to observe output, and in visualizing the state of the graph, and rendering that graph in dot to see it's visualization.

Install Graphviz Before we jump right in, let me strongly encourage you install Graphviz. Graphviz is open source graph visualization software package that includes a handy dandy command line tool call `dot`. Dot is also a standardized and open graph description language that is a key primitive of Graphviz. The `dot` tool in Graphviz takes dot code and renders it nicely. Graphviz is super easy to install. In Ubuntu simply type `sudo apt install graphviz`, or on mac type `brew install graphviz` and you're done! You should be able to call `dot` from the command line.

Ok, lets start with a scenario. Say you'd like to write your first Jac program which will include some nodes, edges, and walkers and you'd like to print to standard output and see what the graph looks like after you run an interesting walker. Let role play.

Lets hop into a `jsctl` shell.

```
haxor@linux:~/jaseci# jsctl -m
Starting Jaseci Shell...
jaseci >
```

Good, we're in! And we've set the session to be an in-memory session so no session file will be created or saved. For this play session we only care about the Jac program we write, which will be saved. The state of the Jaseci machine we run our toy program on doesn't really matter to us.

Now that we've got our shell running, we first want to create a blank graph. Remember, all walkers, Jaseci's primary unit of computation, must run on a node. As default, we can use the root node of a freshly created graph, hence we need to create a base graph. But oh no! We're a bit rusty and have forgotten how create our initial graph using `jsctl`. Let's navigate the help menu to jog our memories.

```
jaseci > help

Documented commands (type help <topic>):
=====
alias check dev graph ls sentinel
architype config edit login object walker

Undocumented commands:
=====
exit help quit

jaseci > help graph
Usage: graph [OPTIONS] COMMAND [ARGS]...

  Group of 'graph' commands

Options:
  --help Show this message and exit.

Commands:
  active Group of 'graph active' commands
  create Create a graph instance and return root node graph object
  delete Permanently delete graph with given id
  get Return the content of the graph with format Valid modes:...
  list Provide complete list of all graph objects (list of root node...)
  node Group of 'graph node' commands
jaseci > graph create --help
Usage: graph create [OPTIONS]

  Create a graph instance and return root node graph object

Options:
  -o, --output TEXT Filename to dump output of this command call.
  -set_active BOOLEAN
  --help Show this message and exit.
jaseci >
```

Ohhh yeah! That's it. After simply using `help` from the shell we were able to navigate to the relevant info for `graph create`. Let's use this newly gotten wisdom.

```
jaseci > graph create -set_active true
{
  "context": {},
  "anchor": null,
  "name": "root",
  "kind": "generic",
  "jid": "urn:uuid:7aa6caff-7a46-4a29-a3b0-b144218312fa",
  "j_timestamp": "2021-08-15T21:34:31.797494",
  "j_type": "graph"
}
jaseci >
```

Great! With this command a graph is created and a single root node is born. `jsctl` shares with us the details of this root graph node. In Jaseci, graphs are referenced by their root nodes and every graph has a single root node.

Notice we've also set the `-set_active` parameter to true. This parameter informs Jaseci to use the root node of this graph (in particular the UUID of this root node) as the default parameter to all future calls to Jaseci Core APIs that have a parameter specifying a graph or node to operate on. This global designation that this graph is the 'active' graph is a convenience feature so we the user doesn't have to specify this parameter for future calls. Of course this can be overridden, more on that later.

Next, let's write some Jac code for our little program. `jsctl` has a built in editor that is simple yet powerful. You can use either this built in editor, or your favorite editor to create the `.jac` file for our toy program. Let's use the built in editor.

```
jaseci > edit fam.jac
```

The `edit` command invokes the built in editor. Though it's a terminal editor based on `ncurses`, you can basically use it much like you'd use any wysiwyg editor with features like standard cut `ctrl-c` and paste `ctrl-v`, mouse text selection, etc. It's based on the phenomenal pure python project from Google called `ci_edit`. For more detailed help cheat sheet see Appendix. If you must use your own favorite editor, simply be sure that you save the `fam.jac` file in the same working directory from which you are running the Jaseci shell. Now type out the toy program in Jac Code 5.2.

```
Jac Code 5.2: Jac Family Toy Program
1  node man;
2  node woman;
3
4  edge mom;
5  edge dad;
6  edge married;
7
8  walker create_fam {
9      root {
10         spawn here --> node::man;
11         spawn here --> node::woman;
12         --> node::man <-[married]-> --> node::woman;
13         take -->;
14     }
15     woman {
16         son = spawn here <-[mom]- node::man;
17         son -[dad]-> <-[married]->;
18     }
19     man {
20         std.out("I_didn't_do_any_of_the_hard_work.");
21     }
22 }
```

Don't worry if that looks like the most cryptic gobbledygook you've ever seen in your life. As you learn the Jac language, all will become clear. For now, lets tinker around. Now save and quit the editor. If you are using the built in editor thats simply a `ctrl-s`, `ctrl-q` combo.

Ok, now we should have a `fam.jac` file saved in our working directory. We can check from the Jaseci shell!

```
jaseci > ls
fam.jac
jaseci >
```

We can list files from the shell prompt. By default the `ls` command only lists files relevant to Jaseci (i.e., `*.jac`, `*.dot`, etc). To list all files simply add a `--all` or `-a`.

Now, on to what is on of the key operations. Lets “register” a sentinel based on our Jac program. A sentinel is the abstraction Jaseci uses to encapsulate compiled walkers and architype nodes and edges. You can think of registering a sentinel as compiling your jac program. The walkers of a given sentinel can then be invoked and run on arbitrary nodes of any graph. Let's register our Jac toy program.

```
jaseci > sentinel register -name fam -code fam.jac -set_active true
2021-08-15 18:03:38,823 - INFO - parse_jac_code: fam: Processing Jac code...
2021-08-15 18:03:39,001 - INFO - register_code: fam: Successfully registered code
{
  "name": "fam",
  "kind": "generic",
  "jid": "urn:uuid:cfc9f017-cb6c-4d06-bc45-758289c96d3f",
  "j_timestamp": "2021-08-15T22:03:38.823651",
  "j_type": "sentinel"
}
jaseci >
```

Ok, theres a lot that just happened there. First, we see some logging output that informs us that the Jac code is being processed (which really means the Jac program is being parsed and IR being generated). If there are any syntax errors or other issues, this is where the error output will be printed along with any problematic lines of code and such. If all goes well, we see the next logging output that the code has been successfully registered. The formal output is the relevant details of the successfully created sentinel. Note, that we've also made this the “active” sentinel meaning it will be used as the default setting for any calls to Jaseci Core APIs that require a sentinel be specified. At this point, Jaseci has registered our code and we are ready to run walkers!

But first, lets take a quick look at some of the objects loaded into our Jaseci machine. For this I'll briefly introduce the `alias` group of APIs.

```
jaseci > alias list
{
  "sentinel:fam": "urn:uuid:cfc9f017-cb6c-4d06-bc45-758289c96d3f",
  "fam:walker:create_fam": "urn:uuid:17598be7-e14f-4000-9d85-66b439fa7421",
  "fam:architype:man": "urn:uuid:c366518d-3b1e-41a3-b1ba-0b9a3ce6e1d6",
  "fam:architype:woman": "urn:uuid:7eb1c510-73ca-49eb-96aa-34357f77b4cb",
  "fam:architype:mom": "urn:uuid:8c9d2a66-4954-4d11-8109-a36b961eeea1",
  "fam:architype:dad": "urn:uuid:d80111e4-62e2-4694-bfaa-f3294d9520d8",
  "fam:architype:married": "urn:uuid:dc4974df-ea57-406e-9468-a1aa5260d306"
}
jaseci >
```

The `alias` set of APIs are designed as an additional set of convenience tools to simplify the referencing of various objects (walkers, architypes, etc) in Jaseci. Instead of having to use the UUIDs to reference each object, an alias can be used to refer to any object. These aliases can be created or removed utilizing the `alias` APIs.

Upon registering a sentinel, a set of aliases are automatically created for each object produced from processing the corresponding Jac program. The call to `alias list` lists all available aliases in the session. Here, we're using this call to see the objects that were created for our toy program and validate it corresponds to the ones we would expect from the Jac Program represented in JC 5.2. Everything looks good!

Now, for the big moment! lets run our walker on the root node of the graph we created and see what happens!

```
jaseci > walker run -name create_fam
I didn't do any of the hard work.
[]
jaseci >
```

Sweet!! We see the standard output we'd expect from our toy program. Hrm, as we'd expect, when it comes to the family, the man doesn't do much it seems.

But there were many semantics to what our toy program does. How do we visualize that the graph produced by our program is right. Well we're in luck! We can use Jaseci 'dot' features to take a look at our graph!!

```
jaseci > graph get -mode dot -o fam.dot
strict digraph root {
  "n0" [ id="550ce1bb405c4477947e019d1e8428eb", label="n0:root" ]
  "n1" [ id="e5c0a9b28f134313a28794a0c061bff1", label="n1:man" ]
  "n2" [ id="bc2d2f18e2de4190a50bec2a32392a4f", label="n2:woman" ]
  "n3" [ id="92ed7781c6674824905b149f7f320fcd", label="n3:man" ]
  "n1" -> "n3" [ id="76535f6c3f0e4b7483c31863299e2784", label="e0:dad" ]
  "n3" -> "n2" [ id="6bb83ee19f8b4f7eb93a11f5d4fa7f0a", label="e1:mom" ]
  "n1" -> "n2" [ id="0fc3550e75f241ce8d1660860cf4e5c9", label="e2:married", dir="both" ]
  "n0" -> "n2" [ id="03fcfb60667b4631b46ee589d982e1ce", label="e3" ]
  "n0" -> "n1" [ id="d1713ac5792e4272b9b20917b0c3ec33", label="e4" ]
}
[saved to fam.dot]
jaseci >
```

Here we've used the `graph get` core API to get a print out of the graph in dot format. By default `graph get` dumps out a list of all edge and node objects of the graph, however with the `-mode dot` parameter we've specified that the graph should be printed in dot. The `-o` flag specifies a file to dump the output of the command. Note that the `-o` flag for `jsctl` commands only outputs the formal returned data (json payload, or string) from a Jaseci Core API. Logging output, standard output, etc will not be saved to the file though anything reported by a walker using `report` will be saved. This output file directive is `jsctl` specific and work with any command given to `jsctl`.




Figure 5.2: Graph for `fam.jac`

To see a pretty visual of the graph itself, we can use the `dot` command from Graphviz. Simply type `dot -Tpdf fam.dot -o fam.pdf` and Voila! We can see the beautiful graph our toy Jac program has produced on its way to the standard output.

Awesomeness! We are Jac Haxors now!

5.2 Jaseci REST API

5.3 Full Spec of Jaseci Core APIs

5.3.1 Cheatsheet

Interface	Parameters
-----------	------------

walker callback	(nd: jaseci.graph.node.node, wlk: jaseci.actor.walker.walker, key: str, ctx: dict = , _req_ctx: dict = , global_sync: bool = True)
walker summon	(key: str, wlk: jaseci.actor.walker.walker, nd: jaseci.graph.node.node, ctx: dict = , _req_ctx: dict = , global_sync: bool = True)
walker register	(snt: jaseci.actor.sentinel.sentinel = None, code: str = "", encoded: bool = False)
walker get	(wlk: jaseci.actor.walker.walker, mode: str = 'default', detailed: bool = False)
walker set	(wlk: jaseci.actor.walker.walker, code: str, mode: str = 'default')
walker list	(snt: jaseci.actor.sentinel.sentinel = None, detailed: bool = False)
walker delete	(wlk: jaseci.actor.walker.walker, snt: jaseci.actor.sentinel.sentinel = None)
walker spawn create	(name: str, snt: jaseci.actor.sentinel.sentinel = None)
walker spawn delete	(name: str)
walker spawn list	(detailed: bool = False)
walker prime	(wlk: jaseci.actor.walker.walker, nd: jaseci.graph.node.node = None, ctx: dict = , _req_ctx: dict =)
walker execute	(wlk: jaseci.actor.walker.walker, prime: jaseci.graph.node.node = None, ctx: dict = , _req_ctx: dict = , profiling: bool = False)
walker run	(name: str, nd: jaseci.graph.node.node = None, ctx: dict = , _req_ctx: dict = , snt: jaseci.actor.sentinel.sentinel = None, profiling: bool = False)
alias register	(name: str, value: str)
alias list	()
alias delete	(name: str)
alias clear	()
global get	(name: str)
global set	(name: str, value: str)
global delete	(name: str)
global sentinel set	(snt: jaseci.actor.sentinel.sentinel = None)
global sentinel unset	()
object get	(obj: jaseci.element.element.element, depth: int = 0, detailed: bool = False)
object perms get	(obj: jaseci.element.element.element)
object perms set	(obj: jaseci.element.element.element, mode: str)
object perms default	(mode: str)
object perms grant	(obj: jaseci.element.element.element, mast: jaseci.element.element.element, read_only: bool = False)
object perms revoke	(obj: jaseci.element.element.element, mast: jaseci.element.element.element)
graph create	(set_active: bool = True)

graph get	(gph: jaseci.graph.graph.graph = None, mode: str = 'default', detailed: bool = False)
graph list	(detailed: bool = False)
graph active set	(gph: jaseci.graph.graph.graph)
graph active unset	()
graph active get	(detailed: bool = False)
graph delete	(gph: jaseci.graph.graph.graph)
graph node get	(nd: jaseci.graph.node.node, ctx: list = None)
graph node set	(nd: jaseci.graph.node.node, ctx: dict, snt: jaseci.actor.sentinel.sentinel = None)
graph walk	(nd: jaseci.graph.node.node = None)
sentinel register	(name: str = 'default', code: str = "", code_dir: str = './', mode: str = 'default', encoded: bool = False, auto_run: str = 'init', auto_gen_graph: bool = True, ctx: dict = , set_active: bool = True)
sentinel pull	(set_active: bool = True, on_demand: bool = True)
sentinel get	(snt: jaseci.actor.sentinel.sentinel = None, mode: str = 'default', detailed: bool = False)
sentinel set	(code: str, code_dir: str = './', encoded: bool = False, snt: jaseci.actor.sentinel.sentinel = None, mode: str = 'default')
sentinel list	(detailed: bool = False)
sentinel test	(snt: jaseci.actor.sentinel.sentinel = None, detailed: bool = False)
sentinel active set	(snt: jaseci.actor.sentinel.sentinel)
sentinel active unset	()
sentinel active global	(detailed: bool = False)
sentinel active get	(detailed: bool = False)
sentinel delete	(snt: jaseci.actor.sentinel.sentinel)
wapi	(name: str, nd: jaseci.graph.node.node = None, ctx: dict = , _req_ctx: dict = , snt: jaseci.actor.sentinel.sentinel = None, profiling: bool = False)
architype register	(code: str, encoded: bool = False, snt: jaseci.actor.sentinel.sentinel = None)
architype get	(arch: jaseci.actor.architype.architype, mode: str = 'default', detailed: bool = False)
architype set	(arch: jaseci.actor.architype.architype, code: str, mode: str = 'default')
architype list	(snt: jaseci.actor.sentinel.sentinel = None, detailed: bool = False)
architype delete	(arch: jaseci.actor.architype.architype, snt: jaseci.actor.sentinel.sentinel = None)
master create	(name: str, set_active: bool = True, other_fields: dict =)
master get	(name: str, mode: str = 'default', detailed: bool = False)
master list	(detailed: bool = False)
master active set	(name: str)

master active unset	()
master active get	(detailed: bool = False)
master self	(detailed: bool = False)
master delete	(name: str)
master createsuper	(name: str, set_active: bool = True, other_fields: dict =)
master allusers	(num: int = 0, start_idx: int = 0)
master become	(mast: jaseci.element.master.master)
master unbecome	()
config get	(name: str, do_check: bool = True)
config set	(name: str, value: str, do_check: bool = True)
config list	()
config index	()
config exists	(name: str)
config delete	(name: str, do_check: bool = True)
logger http connect	(host: str, port: int, url: str, log: str = 'all')
logger http clear	(log: str = 'all')
logger list	()
stripe product create	(name: str = 'VIP Plan', description: str = 'Plan description')
stripe product price set	(productId: str, amount: float = 50, interval: str = 'month')
stripe product list	(detailed: bool = True)
stripe customer create	(paymentId: str, name: str = 'crisopher evangelista', email: str = 'imurbatman12@gmail.com', description: str = 'Myca subscriber')
stripe customer get	(customerId: str)
stripe customer payment add	(paymentMethodId: str, customerId: str)
stripe customer payment delete	(paymentMethodId: str)
stripe customer payment get	(customerId: str)
stripe customer payment default	(customerId: str, paymentMethodId: str)
stripe subscription create	(paymentId: str, name: str, email: str, priceId: str, customerId: str)
stripe subscription delete	(subscriptionId: str)
stripe subscription get	(customerId: str)
stripe invoices list	(customerId: str, subscriptionId: str, limit: int = 10, lastItem: str = ")
actions load local	(file: str)
actions load remote	(url: str)
actions load module	(mod: str)
actions list	(name: str = ")
jac build	(file: str, out: str = ")
jac test	(file: str, detailed: bool = False)
jac run	(file: str, walk: str = 'init', ctx: dict = , profiling: bool = False)
jac dot	(file: str, walk: str = 'init', ctx: dict = , detailed: bool = False)

Table 5.1: Full set of core Jaseci APIs

Part II

The Jac Programming Language

Chapter 6

Jac Language Overview and Basics

To articulate the sorcerer spells made possible by the wand that is Jaseci, I bestow upon thee, the Jac programming language. (Like the Harry Potter [6] simile there? Cool, I know ;-))

The name Jac take was chosen for a few reasons.

- “Jac” is three characters long, so its well suited for the file name extension `.jac` for Jac programs.
- It pulls its letters from the phrase **JA**seci **C**ode.
- And it sounds oh so sweet to say “Did you grok that sick Jac code yet!” Rolls right off the tongue.

This chapter provides the full deep dive into the language. By the end, you will be fully empowered with Jaseci wizardry and get a view into the key insights and novelty in the coding style.

First lets quickly dispense with the mundane. This section covers the standard table stakes fodder present in pretty much all languages. These aspects of Jac must be covered for completeness, however you should be able to speed read this section. If you are unable to speed read this, perhaps you should give visual basic a try.



Figure 6.1: World’s youngest coder with valid HTML on shirt.¹

6.1 The Obligatory Hello World

Let’s begin with what has become the unofficial official starting point for any introduction to a new language, the “hello world” program. Thank you Canada for providing one of the most impactful contributions in computer science with “hello world” becoming a meme both technically and socially. We have such love for this contribution we even tag or newborns with the phrase as per Fig. 6.1. I digress. Lets now christen our baby, Jaseci, with its “Hello World” expression.

```
Jac Code 6.1: Jaseci says Hello!
```

```
1 walker init {  
2   std.out("Hello_World");  
3 }
```

Simple enough right? Well let’s walk through it. What we have here is a valid Jac program with a single walker defined. Remember a walker is our little robot friend that walks the nodes and edges of a graph and does stuff. In the curly braces, we articulate what our walker should do. Here we instruct our walker to utilize the standard library to call a print function denoted as `std.out` to print a single string, our star and esteemed string, “Hello World.” The output to the screen (or wherever the OS is routing it’s standard stream output) is simply,

```
Hello World
```

¹Image credit to wiki contributor [1]

And there we have the most useless program in the world. Though... technically this program is AI. Its not as intelligent as the machine depicted in Figure 6.1, but one that we can understand much better (unless you speak “goo goo gaa gaa” of course). Let’s move on.

6.2 Numbers, Arithmetic, and Logic

6.2.1 Basic Arithmetic Operations

Next we should cover the he simplest math operations in Jac. We build upon what we’ve learned so far with our conversational AI above.

Jac Code 6.2: Basic arithmetic operations

```
1 walker init {  
2   a = 4 + 4;  
3   b = 4 * -5;  
4   c = 4 / 4; # Evaluates to a floating point number  
5   d = 4 - 6;  
6   e = a + b + c + d;  
7   std.out(a, b, c, d, e);  
8 }
```

The output of this groundbreaking program is,

```
8 -20 1.0 -2 -13.0
```

Jac Code 6.2 is comprised of basic math operations. The semantics of these expressions are pretty much the same as anything you may have seen before, and pretty much match the semantics we have in the Python language. In this Example, we also observe that Jac is an untyped language and variables can be declared via a direct assignment; also very Python’y. The comma separated list of the defined variables `a` - `e` in the call to `std.out` illustrate multiple values being printed to screen from a single call.

Additionally, Jac supports power and modulo operations.

Jac Code 6.3: Additional arithmetic operations

```
1 walker init {  
2   a = 4 ^ 4; b = 9 % 5; std.out(a, b);  
3 }
```

Jac Code 6.3 outputs,

256 4

Here, we can also observe that, unlike Python, whitespace does not matter whatsoever. Languages utilizing whitespace to express static scoping should be criminalized. Yeah, I said it, see Rant B.1. Anyway, A corollary to this design decision is that every statement must end with a “;”. The wonderful ;, A nod of respect goes to `C/C++/JavaScript` for bringing this beautiful code punctuation to the masses. Of course the ; as code punctuation was first introduced with `ALGOL 58`, but who the heck knows that language. It sounds like some kind of plant species. Bleh. Onwards.

Nerd Alert 4 *(time to let your eyes glaze over)*

Grammar 6.4 shows the lines from the formal grammar for Jac that corresponds to the parsing of arithmetic.

Grammar 6.4: Jac grammar clip relevant to arithmetic

```

125 arithmetic: term ((PLUS | MINUS) term)*;
126
127 term: factor ((MUL | DIV | MOD) factor)*;
128
129 factor: (PLUS | MINUS) factor | power;
130
131 power: func_call (POW factor)*;

```

(full grammar in Appendix C)

6.2.2 Comparison, Logical, and Membership Operations

Next we review the comparison and logical operations supported in Jac. This is relatively straight forward if you’ve programmed before. Let’s summarize quickly for completeness.

Jac Code 6.5: Comparision operations

```

1  walker init {
2      a = 5; b = 6;
3      std.out(a == b,
4          a != b,
5          a < b,
6          a > b,
7          a <= b,
8          a >= b,
9          a == b-1);
10 }

```

```
false true true false true false true
```

In order of appearance, we have tests for equality, non equality, less than, greater than, less than or equal, and greater than or equal. These tools prove indispensable when expressing functionality through conditionals and loops. Additionally,

Jac Code 6.6: Logical operations

```
1 walker init {
2     a = true; b = false;
3     std.out(a,
4         !a,
5         a && b,
6         a || b,
7         a and b,
8         a or b,
9         !a or b,
10        !(a and b));
11 }
```

```
true false false true false true false true
```

Jac Code 6.6 presents the logical operations supported by Jac. In order of appearance we have, boolean complement, logical and, logical or, another way to express and and or (thank you Python) and some combinations. These are also indispensable when using conditionals.

[NEED EXAMPLE FOR MEMBERSHIP OPERATIONS]

Nerd Alert 5 *(time to let your eyes glaze over)*

Grammar 6.7 shows the lines from the formal grammar for Jac that corresponds to the parsing of comparison, logical, and membership operations.

Grammar 6.7: Jac grammar clip relevant to comparison, logic, and membership

```
117 logical: compare ((KW_AND | KW_OR) compare)*;
118
119 compare: NOT compare | arithmetic (cmp_op arithmetic)*;
120
121 cmp_op: EE | LT | GT | LTE | GTE | NE | KW_IN | nin;
122
123 nin: NOT KW_IN;
```

(full grammar in Appendix C)

6.2.3 Assignment Operations

Next, let's take a look at assignment in Jac. In contrast to equality tests of `==`, assignment operations copy the value of the right hand side of the assignment to the variable or object on the left hand side.

Jac Code 6.8: Assignment operations

```
1  walker init {  
2      a = 4 + 4; std.out(a);  
3      a += 4 + 4; std.out(a);  
4      a -= 4 * -5; std.out(a);  
5      a *= 4 / 4; std.out(a);  
6      a /= 4 - 6; std.out(a);  
7  
8      # a := here; std.out(a);  
9      # Noting existence of copy assign, described later  
10 }
```

```
8  
16  
36  
36.0  
-18.0
```

As shown in Jac Code 6.8, there are a number of ways we can articulate an assignment. Of course we can simply set a variable equal to a particular value, however, we can go beyond that to set that assignment relative to its original value. In particular, we can use the short hand `a += 4 + 4;` to represent `a = a + 4 + 4;`. We will describe later an additional assignment type we call the copy assign. If you're simply dying of curiosity, I'll throw you a bone. This `:=` assignment only applies to nodes and edges and has the semantic of copying the member values of a node or edge as opposed to the particular node or edge a variable is pointing to. In a nutshell this assignment uses pass by value semantics vs pass by reference semantics which is default for nodes and edges.

Rank	Symbol	Description
1	(), [], ., ::, spawn	Parenthetical/grouping, node/edge manipulation
2	^, []	Exponent, Index
3	*, /, %	Multiplication, division, modulo
4	+, -	Addition, subtraction
5	==, !=, >=, <=, >, <, in, not in	Comparison
6	&&, , and, or	Logical
7	-->, <--, -[]->, <-[]-	Connect
8	=, +=, -=, *=, /=, :=	Assignment

Table 6.1: Precedence of operations in Jac

Nerd Alert 6 (*time to let your eyes glaze over*)

Grammar 6.9 shows the lines from the formal grammar for Jac that corresponds to the parsing of assignment operations.

Grammar 6.9: Jac grammar clip relevant to assignment

```

107 expression: connect (assignment | copy_assign | inc_assign)?;
108
109 assignment: EQ expression;
110
111 copy_assign: CPY_EQ expression;
112
113 inc_assign: (PEQ | MEQ | TEQ | DEQ) expression;
```

(full grammar in Appendix C)

6.2.4 Precedence

At this point in our discussion its important to note the precedence of operations in Jac. Table 6.1 summarizes this precedence. There are a number of new and perhaps interesting things that appear in this table that you may not have seen before. [JOKE] For now, don't hurt yourself trying to understand what they are and mean, we'll get there.

6.2.5 Primitive Types

Jac Code 6.10: Primitive types

```

1 walker init {
2   a=5;
3   std.out(a.type, '->', a);
4   a=5.0;
5   std.out(a.type, '->', a);
```

```

6   a=true;
7   std.out(a.type, '-', a);
8   a=[5];
9   std.out(a.type, '-', a);
10  a='5';
11  std.out(a.type, '-', a);
12  a={'num': 5};
13  std.out(a.type, '-', a);
14 }

```

```

JAC_TYPE.INT - 5
JAC_TYPE.FLOAT - 5.0
JAC_TYPE.BOOL - true
JAC_TYPE.LIST - [5]
JAC_TYPE.STR - 5
JAC_TYPE.DICT - {"num": 5}

```

6.2.5.1 Integers and Floats

6.2.5.2 Booleans

6.2.5.3 Lists and Strings

6.2.5.4 Dictionaries

6.2.5.5 Nodes and Edges

Jac Code 6.11: Basic arithmetic operations

```

1  walker init {
2      nd = spawn here --> node::generic;
3      std.out(nd.type, nd);
4      std.out(nd.edge.type, nd.edge);
5      std.out(nd.edge[0].type, nd.edge[0]);
6  }

```

```

JAC_TYPE.NODE jac:uuid:918900e4-9a35-4771-bce8-e1330d761bf6
JAC_TYPE.LIST ["jac:uuid:2930cfd6-7007-4942-b6ab-f28986819336"]
JAC_TYPE.EDGE jac:uuid:2930cfd6-7007-4942-b6ab-f28986819336

```

6.2.5.6 Specials

Jac Code 6.12: Basic arithmetic operations

```

1  walker init {
2      a=null;
3      std.out(a.type, '-', a);
4      a=str;
5      std.out(a.type, '-', a);
6      std.out(null.type);
7      std.out(null.type.type);
8  }

```

```

JAC_TYPE.NULL - null
JAC_TYPE.TYPE - JAC_TYPE.STR
JAC_TYPE.NULL
JAC_TYPE.TYPE

```

[Type type]

[Null]

6.2.5.7 Typecasting

Jac Code 6.13: Basic arithmetic operations

```

1  walker init {
2      a=5.6;
3      std.out(a+2);
4      std.out((a+2).int);
5      std.out((a+2).str);
6      std.out((a+2).bool);
7      std.out((a+2).int.float);
8
9      if(a.str.type == str and !(a.int.type == str) and a.int.type == int):
10         std.out("Types comes back correct");
11  }

```

```

7.6
7
7.6
true
7.0
Types comes back correct

```

6.3 Foreshadowing Unique Graph Operations

Before we move on to more mundane basics that will continue to neutralize any kind of caffeine or methamphetamine buzz an experienced coder might have as they read this, lets enjoy a Jaseci jolt!

As described before, all data in Jaseci lives in either a graph, or within the scope of a walker. A walker, executes when it is *engaged* to the graph, meaning it is located on a particular node of the graph. In the case of the Jac programs we've looked at so far, each program has specified one walker for which I've happened to choose the name `init`. By default these `init` walkers are invoked from the default root node of an empty graph. Figure 6.2 shows the complete state of memory for all of the Jac programs discussed thus far. The `init` walker in these cases does not *walk* anywhere and has only executed a set of operations on this default root node `n0`.

Let's have a quick peek at some slick language syntax for building this graph and traveling to new nodes.

Jac Code 6.14: Preview of graph operators

```

1 node simple;
2 edge back;
3
4 walker kewl_graph_creator {
5     node_a = spawn here --> node::simple;
6     here <-[back]- node_a;
7     node_b = spawn here <--> node::simple;
8     node_b --> node_a;
9 }

```

Jac Code 6.14 presents a sequence of operations that creates nodes and edges and produces a relatively simple complex graph. There is a bunch of new syntactic goodness presented in less than 10 lines of code and I certainly won't describe them all here. The goal is to simply whet your appetite on whats to come. But lets look at the state of our data (memory) shown in Figure 6.3.

Yep, there a good bit going on here. in less than 10 lines of code we've done the following things:

1. Specified a new type of node we call a simple node.
2. Specified a new type of edge we call a back edge.
3. Specified a walker `kewl_graph_creator` and its behavior
4. Instantiated a outward pointing edge from the `n0:root` node.




Figure 6.2: Graph in memory for simple Hello World program (JC 6.1)




Figure 6.3: Graph in memory for JC 6.14

5. Instantiated an instance of node type `simple`
6. Connected edge from from `root` to `n1`
7. Instantiated a `back` edge
8. Connected `back` edge from `n1` to `n0`
9. Instantiated another instance of node type `simple`, `n2`
10. Instantiated an undirected edge from the `n0:root` node.
11. Connected edge from `root` to `n2`
12. Instantiated an outward pointing edge from `n2`
13. Connected edge from `n2` to `n1`

Don't worry, I'll wait till that sinks in... Good? Well, if you liked that, just you wait.

This is going to get very interesting indeed, but first, on to more standard stuff...

6.4 More on Strings, Lists, and Dictionaries

Jac Code 6.15: Built-in String Library

```

1  walker init {
2      a="␣tEsting␣me␣␣";
3      report a[4];
4      report a[4:7];
5      report a[3:-1];
6      report a.str::upper;
7      report a.str::lower;
8      report a.str::title;
9      report a.str::capitalize;
10     report a.str::swap_case;
11     report a.str::is_alnum;
12     report a.str::is_alpha;
13     report a.str::is_digit;
14     report a.str::is_title;
15     report a.str::is_upper;
16     report a.str::is_lower;
17     report a.str::is_space;
18     report '{"a":␣5}' .str::load_json;
19     report a.str::count('t');
20     report a.str::find('i');
21     report a.str::split;
22     report a.str::split('E');
23     report a.str::startswith('tEs');
24     report a.str::endswith('me');
25     report a.str::replace('me', 'you');
26     report a.str::strip;
27     report a.str::strip('␣t');
28     report a.str::lstrip;
29     report a.str::lstrip('␣tE');

```

```
30 report a.str::rstrip;
31 report a.str::rstrip('ue');
32
33 report a.str::upper.str::is_upper;
34 }
```

```
{
  "success": true,
  "report": [
    "t",
    "tin",
    "sting me ",
    " TESTING ME ",
    " testing me ",
    " Testing Me ",
    " testing me ",
    " TeSTING ME ",
    false,
    false,
    false,
    false,
    false,
    false,
    false,
    2,
    5,
    [
      "tEsting",
      "me"
    ],
    [
      " t",
      "sting me "
    ],
    false,
    false,
    " tEsting you ",
    "tEsting me",
    "Esting me",
    "tEsting me ",
    "sting me ",
    " tEsting me",
    " tEsting m",
    true
  ]
}
```

Op	Args	Description
<code>.str::upper</code>	none	
<code>.str::lower</code>	none	
<code>.str::title</code>	none	
<code>.str::capitalize</code>	none	
<code>.str::swap_case</code>	none	
<code>.str::is_alnum</code>	none	
<code>.str::is_digit</code>	none	
<code>.str::is_title</code>	none	
<code>.str::is_upper</code>	none	
<code>.str::is_lower</code>	none	
<code>.str::is_space</code>	none	
<code>.str::load_json</code>	none	
<code>.str::count</code>	(substr, start, end)	Returns the number of occurrences of a substring in the given string. Start and end specify range of indices to search
<code>.str::find</code>	(substr, start, end)	Returns the index of first occurrence of the substring (if found). If not found, it returns -1. Start and end specify range of indices to search.
<code>.str::split</code>	optional (separator, maxsplit)	Breaks up a string at the specified separator for maxsplit number of times and returns a list of strings. Default separator is ' ' and maxsplit is unlimited.
<code>.str::startswith</code>		
<code>.str::endswith</code>		
<code>.str::replace</code>		
<code>.str::strip</code>	optional,	
<code>.str::lstrip</code>	optional,	
<code>.str::rstrip</code>	optional,	

Table 6.2: String operations in Jac

6.4.1 Library of String Operations

6.4.2 Library of List Operations

6.4.3 Library of Dictionary Operations

6.5 Control Flow

Jac Code 6.16: if statement

```

1 walker init {
2   a = 4; b = 5;
3   if(a < b): std.out("Hello!");
4 }
```

Op	Args	Description
<code>.list::max</code>	none	
<code>.list::min</code>	none	
<code>.list::idx_of_max</code>	none	
<code>.list::idx_of_min</code>	none	
<code>.list::copy</code>	none	Returns a shallow copy of the list
<code>.list::deepcopy</code>	none	Returns a deep copy of the list
<code>.list::sort</code>	none	
<code>.list::reverse</code>	none	
<code>.list::clear</code>	none	
<code>.list::pop</code>	optional,	
<code>.list::index</code>		
<code>.list::append</code>		
<code>.list::extend</code>		
<code>.list::insert</code>		
<code>.list::remove</code>		
<code>.list::count</code>		

Table 6.3: List operations in Jac

Op	Args	Description
<code>.dict::items</code>	none	
<code>.dict::copy</code>	none	Returns a shallow copy of the dictionary
<code>.dict::deepcopy</code>	none	Returns a deep copy of the dictionary
<code>.dict::keys</code>	none	
<code>.dict::clear</code>	none	
<code>.dict::popitem</code>	none	
<code>.dict::values</code>	none	
<code>.dict::pop</code>		
<code>.dict::update</code>		

Table 6.4: Dictionary operations in Jac

```
Hello!
```

```
Jac Code 6.17: else statement
1 walker init {
2   a = 4; b = 5;
3   if(a == b): std.out("A_equals_B");
4   else: std.out("A_is_not_equal_to_B");
5 }
```

```
A is not equal to B
```

```
Jac Code 6.18: elif statement
1 walker init {
```

```
2  a = 4; b = 5;
3  if(a == b): std.out("A_equals_B");
4  elif(a > b): std.out("A_is_greater_than_B");
5  elif(a == b - 1): std.out("A_is_one_less_than_B");
6  elif(a == b - 2): std.out("A_is_two_less_than_B");
7  else: std.out("A_is_something_else");
8  }
```

A is one less than B

Jac Code 6.19: for loop

```
1  walker init {
2      for i=0 to i<10 by i+=1:
3          std.out("Hello", i, "times!");
4  }
```

Hello 0 times!
Hello 1 times!
Hello 2 times!
Hello 3 times!
Hello 4 times!
Hello 5 times!
Hello 6 times!
Hello 7 times!
Hello 8 times!
Hello 9 times!

Jac Code 6.20: for loop through list

```
1  walker init {
2      my_list = [1, 'jon', 3.5, 4];
3      for i in my_list:
4          std.out("Hello", i, "times!");
5  }
```

Hello 1 times!
Hello jon times!
Hello 3.5 times!
Hello 4 times!

Jac Code 6.21: while loop

```
1  walker init {
2      i = 5;
3      while(i>0) {
4          std.out("Hello", i, "times!");
5          i -= 1;
6      }
```

```
6   }  
7 }
```

```
Hello 5 times!  
Hello 4 times!  
Hello 3 times!  
Hello 2 times!  
Hello 1 times!
```

Jac Code 6.22: break statement

```
1 walker init {  
2     for i=0 to i<10 by i+=1 {  
3         std.out("Hello", i, "times!");  
4         if(i == 6): break;  
5     }  
6 }
```

```
Hello 0 times!  
Hello 1 times!  
Hello 2 times!  
Hello 3 times!  
Hello 4 times!  
Hello 5 times!  
Hello 6 times!
```

Jac Code 6.23: continue statement

```
1 walker init {  
2     i = 5;  
3     while(i>0) {  
4         if(i == 3){  
5             i -= 1; continue;  
6         }  
7         std.out("Hello", i, "times!");  
8         i -= 1;  
9     }  
10 }
```

```
Hello 5 times!  
Hello 4 times!  
Hello 2 times!  
Hello 1 times!
```

Chapter 7

Graphs, Architypes, and Walkers in Jac

7.1 Structure of a Jac Program

[Introduce structure of a jac program]

[Specify the difference between graph architypes, graph instantiations, and walkers]

[Present simple program that utilizes the structures]

[Present variations on articulating the same program]

[Code blocks]

Nerd Alert 7 *(time to let your eyes glaze over)*

Grammar 7.1 shows the lines from the formal grammar for Jac that presents the high level structure of a Jac program.

Grammar 7.1: Jac grammar clip relevant to arithmetic

```

3  start: ver_label? element+ EOF;
4
5  element: archetype | walker;
6
7  archetype:
8      KW_NODE NAME (COLON INT)? attr_block
9      | KW_EDGE NAME attr_block
10     | KW_GRAPH NAME graph_block;
11
12  walker:
13      KW_WALKER NAME namespaces? LBRACE attr_stmt* walk_entry_block? (
14          statement
15          | walk_activity_block
16      ) * walk_exit_block? RBRACE;

```

(full grammar in Appendix C)

7.2 Graphs as First Class Citizens

7.2.1 Connect and Spawn operations

Jac Code 7.2: Simple walker creating and connected nodes

```

1  walker init {
2      node1 = spawn node::generic;
3      node2 = spawn node::generic;
4      node1 <--> node2;
5      here --> node1;
6      node2 <-- here;
7  }

```

Jac Code 7.3: Creating named node types

```

1  node person;
2  edge family;
3  edge friend;
4
5  walker init {
6      node1 = spawn node::person;
7      node2 = spawn node::person;
8      node1 <-[family]-> node2;

```




Figure 7.1: Graph in memory for JC 7.2




Figure 7.2: Graph in memory for JC 7.3

```

9   here -[friend]-> node1;
10  node2 <-[friend]- here;
11
12  # named and unnamed edges and nodes can be mixed
13  node2 --> here;
14  }

```

Jac Code 7.4: Connecting nodes within spawn statement

```

1  node person;
2  edge friend;
3  edge family;
4
5  walker init {
6    node1 = spawn here -[friend]-> node::person;
7    node2 = spawn node1 <-[family]-> node::person;
8    here -[friend]-> node2;
9  }

```




Figure 7.3: Graph in memory for JC 7.4




Figure 7.4: Graph in memory for JC 7.5

Jac Code 7.5: Chaining node connections using the connect operator

```

1 node person;
2 edge friend;
3 edge family;
4
5 walker init {
6   node1 = spawn node::person;
7   node2 = spawn node::person;
8   node2 <-[friend]- here -[friend]-> node1 <-[family]-> node2;
9 }

```

Another incredibly useful notion to consider about connect operations is that they can be chained. The same graph shown in Figure 7.4 can be achieved with the chained usage of the connect operation in line 8 of JC 7.5. Here nodes are chained in an intuitive left-to-right manor. Relatively sophisticated graph structures can be rapidly expressed using chained connect operations.

7.2.2 Static Graph Creation

7.2.2.1 Static Spawn Graphs

Jac Code 7.6: A Spawn style static graph

```

1 graph hlp_graph {
2   has anchor graph_root;
3   spawn {
4     graph_root = spawn node::state(name="root_state");
5     user_node = spawn node::user;
6
7     state_home_price_inquiry = spawn node::state(name="home_price_inquiry");
8     state_prob_of_approval = spawn node::state(name="prob_of_approval");
9
10    graph_root -[user]-> user_node;
11
12    graph_root -[transition(intent_label = "home_price_inquiry")]->
13      ↪ state_home_price_inquiry;
14    graph_root -[transition(intent_label = "robability_of_loan_approval")]->
15      ↪ state_prob_of_approval;
16    state_home_price_inquiry -[transition(intent_label = "specifying_location")]->
17      ↪ state_home_price_inquiry;
18    state_home_price_inquiry -[transition(intent_label = "home_price_inquiry")]->
19      ↪ state_home_price_inquiry;
20
21    state_home_price_inquiry -[transition(intent_label = "probability_of_loan_approval"
22      ↪ )]-> state_prob_of_approval;
23    state_prob_of_approval -[transition(intent_label = "home_price_inquiry")]->
24      ↪ state_home_price_inquiry;
25  }
26 }

```

Jac Code 7.7: Associated DOT style static graph

```

1 graph acme_graph_dot {
2   has anchor state_conv_root;
3   graph G {
4     state_conv_root [node=conv_state, name=conv_root]
5
6     state_office_hour [node=conv_state, name=office_hour]
7     state_payment_method [node=conv_state, name=payment_method]
8     state_phone_number [node=conv_state, name=phone_number]
9     state_email_address [node=conv_state, name=email_address]
10    state_promotions [node=conv_state, name=promotions]
11
12    state_cancel_appointment [node=conv_state, name=cancel_appointment]
13    state_reschedule_appointment [node=conv_state, name=reschedule_appointment]
14    state_refunds [node=conv_state, name=refunds]
15    state_feedback [node=conv_state, name=feedback]
16
17    state_service_inquiry [node=conv_state, name=service_inquiry]
18
19    state_conv_root -> state_office_hour [edge=transition, intent="office_hour"]
20    state_conv_root -> state_payment_method [edge=transition, intent="payment_method"]
21    state_conv_root -> state_phone_number [edge=transition, intent="phone_number"]
22    state_conv_root -> state_email_address [edge=transition, intent="email_address"]
23    state_conv_root -> state_promotions [edge=transition, intent="promotions"]

```

```

24     state_conv_root -> state_cancel_appointment [edge=transition, intent="cancel_
      ↳ appointment"]
25     state_conv_root -> state_reschedule_appointment [edge=transition, intent="
      ↳ reschedule_appointment"]
26     state_conv_root -> state_refunds [edge=transition, intent="refunds"]
27     state_conv_root -> state_feedback [edge=transition, intent="feedback"]
28     state_conv_root -> state_service_inquiry [edge=transition, intent="service_inquiry"
      ↳ ]
29 }
30 }

```

7.2.2.2 Static DOT Graphs

Jac Code 7.8: A DOT style static graph

```

1  node test_node {
2      has name;
3  }
4  edge special;
5  graph test_graph {
6      has anchor graph_root;
7      graph G {
8          graph_root [node=test_node, name=root]
9          node_1 [node=test_node, name=node_1]
10         node_2 [node=test_node, name=node_2]
11         graph_root -> node_1 [edge=special]
12         graph_root -> node_2
13     }
14 }
15 walker init {
16     has nodes;
17     with entry {
18         nodes = [];
19     }
20     root {
21         spawn here --> graph::test_graph;
22         take --> node::test_node;
23     }
24     test_node {
25         nodes += [here];
26         take -[special]-> node::test_node;
27     }
28     report here;
29 }

```

```

{
  "success": true,
  "report": [
    {
      "context": {},
      "anchor": null,
      "name": "root",
      "kind": "generic",
      "jid": "urn:uuid:0ac65923-90b5-4c10-bda0-65ec6a2c36e7",
      "j_timestamp": "2022-03-21T00:41:16.715258",
      "j_type": "graph"
    },
    {
      "context": {
        "name": "root"
      },
      "anchor": null,
      "name": "test_node",
      "kind": "node",
      "jid": "urn:uuid:60e68110-7a11-446e-a333-57d75d12e7d7",
      "j_timestamp": "2022-03-21T00:41:16.750759",
      "j_type": "node"
    },
    {
      "context": {
        "name": "node_1"
      },
      "anchor": null,
      "name": "test_node",
      "kind": "node",
      "jid": "urn:uuid:fecae690-a50d-4f2c-91e2-e8ec083c5443",
      "j_timestamp": "2022-03-21T00:41:16.750876",
      "j_type": "node"
    }
  ]
}

```

Jac Code 7.9: Another DOT style static graph

```

1  node year {
2    has color;
3  }
4  node month {
5    has count, season;
6  }
7  node week;
8  node day;
9  edge parent;
10 edge child;
11 graph test_graph {
12   has anchor A;
13   strict graph G {
14     H [node=year]
15     C [node=week]

```

```

16     E [node=day]
17     D [node=day]
18
19     A -> B // Basic directional edge
20     B -- H // Basic non-directional edge
21     B -> C [edge=parent] // Edge with attribute
22     C -> D -> E [edge=child] // Chain edge
23
24     A [color=red] // Node with DOT builtin graphing attr
25     B [node=month, count=2] [season=spring] // Node with Jac attr
26     A [node=year] // Multiple attr statement per node
27 }
28 }
29 walker init {
30     root {
31         spawn here --> graph::test_graph;
32     }
33     take -->;
34     report here.details['name'];
35 }

```

```

{
  "success": true,
  "report": [
    "root",
    "year",
    "month",
    "year",
    "week",
    "day",
    "day"
  ]
}

```

7.3 Walkers as the second First Class Citizens

Jac Code 7.10: Walkers spawning other walkers

```

1 node person;
2 edge friend;
3 edge family;
4
5 walker friend_ties {
6     for i in -[friend]->:
7         std.out(here, 'is_related_to\n', i, '\n');
8 }
9
10 walker init {
11     node1 = spawn here -[friend]-> node::person;
12     node2 = spawn node1 <-[family]-> node::person;

```




Figure 7.5: Graph in memory for JC 7.10

```

13   here -[friend]-> node2;
14   spawn here walker::friend_ties;
15 }

```

```

graph:generic:root:urn:uuid:f93bca4a-a722-4fd7-b5e1-55372b4dd314 is related to
node:node:person:urn:uuid:18411a74-60ac-4223-9d59-c3e6a8de7179

```

```

graph:generic:root:urn:uuid:f93bca4a-a722-4fd7-b5e1-55372b4dd314 is related to
node:node:person:urn:uuid:2d251260-3086-4f4f-b5e0-fd36f6043ac7

```

Jac Code 7.11: Getting returned values from spawned walkers

```

1  node person;
2  edge friend;
3  edge family;
4
5  walker friend_ties {
6    has anchor fam_nodes;
7    fam_nodes = -[friend]->;
8  }
9
10 walker init {
11   node1 = spawn here -[friend]-> node::person;
12   node2 = spawn node1 <-[family]-> node::person;
13   here -[friend]-> node2;
14   fam = spawn here walker::friend_ties;
15   for i in fam:
16     std.out(here, 'is_related_to\n', i, '\n');
17 }

```





Figure 7.6: Graph in memory for JC 7.11

```
graph:generic:root:urn:uuid:75d1050b-a010-4e6d-ad6a-c941d5ce57ce is related to
node:node:person:urn:uuid:b1b6ead0-0fc6-4736-928a-f8500832fb3b
```

```
graph:generic:root:urn:uuid:75d1050b-a010-4e6d-ad6a-c941d5ce57ce is related to
node:node:person:urn:uuid:914af4dd-6d5a-4f00-a70c-8871db4a8b95
```

Jac Code 7.12: Increasing elegance by remembering spawns are expressions

```

1  node person;
2  edge friend;
3  edge family;
4
5  walker friend_ties {
6    has anchor fam_nodes;
7    fam_nodes = -[friend]->;
8  }
9
10 walker init {
11   node1 = spawn here -[friend]-> node::person;
12   node2 = spawn node1 <-[family]-> node::person;
13   here -[friend]-> node2;
14   for i in spawn here walker::friend_ties:
15     std.out(here, 'is_related_to\n', i, '\n');
16 }
```

Walkers are entry points to all valid jac programs

7.4 Architypes

7.4.1 Context on Nodes and Edges




Figure 7.7: Graph in memory for JC 7.13

Jac Code 7.13: Binding member contexts to nodes and edges

```

1 node person {
2   has name;
3   has age;
4   has birthday, profession;
5 }
6
7 edge friend: has meeting_place;
8 edge family: has kind;
9
10 walker init {
11   person1 = spawn here -[friend]-> node::person;
12   person2 = spawn here -[family]-> node::person;
13   person1.name = "Josh"; person1.age = 32;
14   person2.name = "Jane"; person2.age = 30;
15   e1 = -[friend]->.edge[0];
16   e1.meeting_place = "college";
17   e2 = -[family]->.edge[0];
18   e2.kind = "sister";
19
20   std.out("Context_for_our_people_nodes:");
21   for i in -->: std.out(i.context);
22   # or, for i in -->.node: std.out(i.context);
23   std.out("\nContext_for_our_edges_to_those_people:");
24   for i in -->.edge: std.out(i.context);
25 }

```

Context for our people nodes:

```
{'name': 'Josh', 'age': 32, 'birthday': '', 'profession': ''}
{'name': 'Jane', 'age': 30, 'birthday': '', 'profession': ''}
```

Context for our edges to those people:

```
{'meeting_place': 'college'}
{'type': 'sister'}
```

Jac Code 7.14: Binding contexts with less code

```

1 node person: has name, age, birthday, profession;
2 edge friend: has meeting_place;
3 edge family: has kind;

```

```

4
5 walker init {
6   person1 = spawn here -[friend(meeting_place = "college")] ->
7     node::person(name = "Josh", age = 32);
8   person2 = spawn here -[family(kind = "sister")] ->
9     node::person(name = "Jane", age = 30);
10
11   std.out("Context for our people nodes and edges:");
12   for i in -->: std.out(i.context, '\n', i.edge[0].context);
13 }

```

```

Context for our people nodes and edges:
{'name': 'Josh', 'age': 32, 'birthday': '', 'profession': ''}
{'meeting_place': 'college'}
{'name': 'Jane', 'age': 30, 'birthday': '', 'profession': ''}
{'type': 'sister'}

```

7.4.2 Copy Assignment Operator

Jac Code 7.15: Copy assigning from node to node

```

1 node person: has name, age, birthday, profession;
2 edge friend: has meeting_place;
3 edge family: has kind;
4
5 walker init {
6   person1 = spawn here -[friend(meeting_place = "college")] ->
7     node::person(name = "Josh", age = 32);
8   person2 = spawn here -[family(kind = "sister")] ->
9     node::person(name = "Jane", age = 30);
10
11   twin1 = spawn here -[friend]-> node::person;
12   twin2 = spawn here -[family]-> node::person;
13   twin1 := person1;
14   twin2 := person2;
15
16   -->.edge[2] := -->.edge[0];
17   -->.edge[3] := -->.edge[1];
18
19   std.out("Context for our people nodes and edges:");
20   for i in -->: std.out(i.context, '\n', i.edge[0].context);
21 }

```




Figure 7.8: Graph in memory for JC 7.15

```

{'name': 'Josh', 'age': 32, 'birthday': '', 'profession': ''}
{'meeting_place': 'college'}
{'name': 'Jane', 'age': 30, 'birthday': '', 'profession': ''}
{'type': 'sister'}
{'name': 'Josh', 'age': 32, 'birthday': '', 'profession': ''}
{'meeting_place': 'college'}
{'name': 'Jane', 'age': 30, 'birthday': '', 'profession': ''}
{'type': 'sister'}

```

7.4.3 Plucking Values from Node and Edge Sets

Another very handy dandy feature when interacting with collections of nodes and edges is to quickly extract a list of all the values for a given `has` variable across the collection of nodes or edges. Lets look at an example.

Jac Code 7.16: Plucking values out of nodes and edges

```

1 node simple: has n_name;
2 edge conn: has e_name;
3
4 walker node_edge_plucking {
5   with entry {
6     for i=0 to i<3 by i+=1:
7       spawn here -[conn(e_name="edge"+i.str)]-> node::simple(n_name="node"+i.str);
8   }
9   std.out(-->.n_name);
10  std.out(-->.edge.e_name);
11 }

```

```

["node0", "node1", "node2"]
["edge0", "edge1", "edge2"]

```

As shown in JC 7.16 we are referencing the `has` variable of the archetypes for the collection of `simple` nodes and `conn` edges on lines 8 and 9 respectively. As can be seen in the output, these references evaluate to a list of the values for the corresponding variables. Keep in mind this can work with a mixture of nodes and edges in a collection given they share a

given `has` variable name.

7.4.4 Referencing and Dereferencing Nodes and Edges

Nodes and edges can be referenced and dereferenced. These operations are synonymous with they way references work in many languages and borrows the syntax of pointers in C/C++. In particular, the `&` is used to get the reference of an object and `*` is used to dereference object. However, in contrast to C/C++, instead of the references representing memory location in word format, references in Jac uses a unique identifier (in UUID format) for the object.

Jac Code 7.17: Rereferences and dereferences in Jac

```

1  node simple: has name;
2
3  walker ref_deref {
4    with entry {
5      for i=0 to i<3 by i+=1:
6        spawn here --> node::simple(name="node"+i.str);
7    }
8    var = &(-->[0]);
9    std.out('ref:', var);
10   std.out('obj:', *var);
11   std.out('info:', (*var).info);
12 }
```

```

ref: urn:uuid:04295f7f-a5bf-4db3-87ce-e13653a81b25
obj: jac:uuid:04295f7f-a5bf-4db3-87ce-e13653a81b25
info: {"context": {"name": "node0"}, "anchor": null, "name": "simple", "kind": "node", "
    ↪ jid": "urn:uuid:04295f7f-a5bf-4db3-87ce-e13653a81b25", "j_timestamp": "2022-08-10
    ↪ T15:57:00.577287", "j_type": "node"}
```

JC 7.17 shows an example of the behavior of references and dereferences in Jac. Note that once dereferenced `var` is simply a UUID formatted string with the unique identifier of the object itself. This UUID is equivalent to the `jid` in the object `.info`. These referencing and dereferencing operations are quite useful for input and output of node locations to a client side, etc.

Nerd Alert 8 *(time to let your eyes glaze over)*

Important Note: The internal representation of an instance of an archetype is a string composed of any UUID that starts with `"jac:uuid:"`. This may change in the future but, if you were to manually assign such a string to a variable in a Jac program, the program will treat this variable like an object.




Figure 7.9: Graph in memory for JC 7.18 and 7.19

7.5 Actions and Abilities

7.5.1 Actions

Jac Code 7.18: Basic action in walker

```

1 node person {
2   has name;
3   has birthday;
4 }
5
6 walker init {
7   can date.quantize_to_year;
8   person1 = spawn here -->
9     node::person(name="Josh", birthday="1995-05-20");
10  birthyear = date.quantize_to_year(person1.birthday);
11  std.out(birthyear);
12 }
  
```

1995-01-01T00:00:00

Jac Code 7.19: Basic action in node

```

1 node person {
2   has name;
3   has birthday;
4   can date.quantize_to_year;
5 }
6
7 walker init {
8   root {
9     person1 = spawn here -->
10      node::person(name="Josh", birthday="1995-05-20");
  
```

```

11     take -->;
12 }
13 person {
14     birthday = date.quantize_to_year(here.birthday);
15     std.out(birthday);
16 }
17 }

```

7.5.2 Fused Interactions Between Nodes and Actions

Jac Code 7.20: Basic action with presets and event triggers

```

1 node person {
2     has name;
3     has byear;
4     can date.quantize_to_year::visitor.year::>byear with setter entry;
5     can std.out::byear,"_from_",visitor.info:: with exit;
6 }
7
8 walker init {
9     has year=std.time_now();
10    root {
11        person1 = spawn here -->
12            node::person(name="Josh", byear="1992-01-01");
13        take --> ;
14    }
15    person {
16        spawn here walker::setter;
17    }
18 }
19
20 walker setter {
21     has year="1995-01-01";
22 }

```

```

1995-01-01T00:00:00 from {'context': {'year': '1995-01-01'}, 'anchor': None, 'name': '
    ↳ setter', 'kind': 'walker', 'jid': 'urn:uuid:6bbf69c3-b95c-4a88-a783-cb793cec4034',
    ↳ 'j_timestamp': '2021-12-04T15:13:13.441516', 'j_type': 'walker'}
1995-01-01T00:00:00 from {'context': {'year': '2021-12-04T15:13:13.440803'}, 'anchor':
    ↳ None, 'name': 'init', 'kind': 'walker', 'jid': 'urn:uuid:7f9d1462-6562-4d4d-ba57-
    ↳ f069c74dfe1e', 'j_timestamp': '2021-12-04T15:13:13.438072', 'j_type': 'walker'}

```

Jac Code 7.21: Basic action with presets and event triggers

```

1 node person {
2     has name;
3     has birthday;
4     can date.quantize_to_year with activity; # <-- walkers can call
5 }

```

```

6
7 walker init {
8   root {
9     person1 = spawn here -->
10      node::person(name="Josh", birthday="1995-05-20");
11     take -->;
12   }
13   person {
14     birthyear = date.quantize_to_year(here.birthday);
15     std.out(birthyear);
16   }
17 }

```

[Only nodes can have with entry/exit“ and presets]

[can leave output (push returns) in node and walker]

```

1995-01-01T00:00:00 from {'context': {'year': '1995-01-01'}, 'anchor': None, 'name': '
  ↳ setter', 'kind': 'walker', 'jid': 'urn:uuid:6bbf69c3-b95c-4a88-a783-cb793cec4034',
  ↳ 'j_timestamp': '2021-12-04T15:13:13.441516', 'j_type': 'walker'}
1995-01-01T00:00:00 from {'context': {'year': '2021-12-04T15:13:13.440803'}, 'anchor':
  ↳ None, 'name': 'init', 'kind': 'walker', 'jid': 'urn:uuid:7f9d1462-6562-4d4d-ba57-
  ↳ f069c74dfele', 'j_timestamp': '2021-12-04T15:13:13.438072', 'j_type': 'walker'}

```

7.5.3 Abilities

Jac Code 7.22: Actions and Abilities in Walkers

```

1 node person {
2   has name;
3   has byear;
4   can set_year with setter entry {
5     byear = visitor.year;
6   }
7   can print_out with exit {
8     std.out(byear, "from", visitor.info);
9   }
10  can reset { #<-- Could add 'with activity' for equivalent behavior
11    ::set_back_to_95;
12    std.out("resetting_year to 1995:", here.context);
13  }
14  can set_back_to_95: byear="1995-01-01";
15 }
16
17 walker init {
18   has year=std.time_now();
19   can setup {
20     person1 = spawn here --> node::person;
21     std.out(person1);
22     person1::reset;

```



```

23   }
24   root {
25       ::setup;
26       take --> ;
27   }
28   person {
29       spawn here walker::setter;
30       person1::reset(name="Joe");
31   }
32 }
33
34 walker setter {
35     has year=std.time_now();
36 }

```

Jac Code 7.23: Abilities in nodes

```

1  node person {
2      has name;
3      has byear;
4      can set_year with setter entry {
5          byear = visitor.year;
6      }
7      can print_out with exit {
8          std.out(byear, "from", visitor.info);
9      }
10     can reset { #<-- Could add 'with activity' for equivalent behavior
11         byear="1995-01-01";
12         std.out("resetting birth year to 1995:", here.context);
13     }
14 }
15
16 walker init {
17     has year=std.time_now();
18     root {
19         person1 = spawn here --> node::person;
20         std.out(person1);
21         person1::reset;
22         take --> ;
23     }
24     person {
25         spawn here walker::setter;
26         here::reset(name="Joe");
27     }
28 }
29
30 walker setter {
31     has year=std.time_now();
32 }

```

7.5.4 here and visitor, the ‘this’ references of Jac

Observe the usage of `here` and `visitor` in the `person` node archetype in JC 7.23. These are synonymous to the `this` reference present in many other languages except `here` point to the current node scope relevant to the execution point in the program and `visitor` points to the relevant walker scope relevant to that given point of execution. These references provide full access to all `has` variables and builtin attributes and operations of the referenced object instance.

Do note that in the context of the `person` node abilities in JC 7.23 a `here` reference to say `here.name = "joe";` would be equivalent to simply `name = "joe";` however to capture the `here.context` (or info/details/etc) the `here` reference becomes quite useful. The similar relationship applies to using `visitor` in walker abilities.

7.6 Inheritance

Chapter 8

Walkers Navigating Graphs

8.1 Taking Edges (and Nodes?)

8.1.1 Basic Walks

Jac Code 8.1: Basic example of walker traveling graph

```
1  node person: has name;
2
3  walker get_names {
4      std.out(here.name);
5      take -->;
6  }
7
8  walker build_example {
9      node1 = spawn here --> node::person(name="Joe");
10     node2 = spawn node1 --> node::person(name="Susan");
11     spawn node2 --> node::person(name="Matt");
12 }
13
14 walker init {
15     root {
16         spawn here walker::build_example;
17         take -->;
18     }
19     person {
20         spawn here walker::get_names;
21         disengage;
22     }
23 }
```




Figure 8.1: Graph in memory for JC 8.3

Jac Code 8.2: Fan out style takes

```

1  node person: has name;
2
3  walker build_example {
4      spawn here -[friend]-> node::person(name="Joe");
5      spawn here -[friend]-> node::person(name="Susan");
6      spawn here -[family]-> node::person(name="Matt");
7  }
8
9  walker init {
10     root {
11         spawn here walker::build_example;
12         take -->;
13     }
14     person {
15         std.out(here.name);
16     }
17 }

```

8.1.2 Breadth First vs Depth First Walks

If you've played with the basic `take` command a bit you would notice that by default it results in a breadth first traversal of a graph. However, the `take` command is indeed quite flexible. You can specify an orientation of the `take` command to navigate with a breadth first or a depth first traversal.

Jac Code 8.3: Breadth first navigation with take vs depth first

```

1  node plain: has name;
2
3  graph example {
4      has anchor head;
5      spawn {
6          n=[];
7          for i=0 to i<7 by i+=1 {
8              n.l::append(spawn node::plain(name=i+1));
9          }
10         n[0] --> n[1] --> n[2];
11             n[1] --> n[3];
12         n[0] --> n[4] --> n[5];
13             n[4] --> n[6];
14         head=n[0];
15     }
16 }
17
18 walker walk_with_breadth {
19     has anchor node_order = [];
20     node_order.l::append(here.name);
21     take:bfs -->; #take:b can also be used
22 }
23
24 walker walk_with_depth {
25     has anchor node_order = [];
26     node_order.l::append(here.name);
27     take:dfs -->; #take:d can also be used
28 }
29
30 walker init {
31     start = spawn here --> graph::example;
32     b_order = spawn start walker::walk_with_breadth;
33     d_order = spawn start walker::walk_with_depth;
34     std.out("Walk_ with_ Breadth:",b_order,"\nWalk_ with_ Depth:",d_order);
35 }

```

Take for example the program shown in JC 8.3. First we observe the definition of a static three level binary tree with the graph `example` on line 3. This is a vanilla structure as depicted in Figure 8.1. Two walkers are present in this example, one walker `walk_with_breadth`, for which we observe a call to `take:bfs -->;` indicating a breadth first traversal, and another walker `walk_with_depth`, for which we observe a call to `take:dfs -->;` indicating a depth first traversal.

As can be seen in its output,

```
Walk with Breadth: [1, 2, 5, 3, 4, 6, 7]
Walk with Depth: [1, 2, 3, 4, 5, 6, 7]
{
  "success": true,
  "report": []
}
```

The print statement on line 34 demonstrate the order of nodes visited correspond to the specified traversal order.

Additionally, the short hand of `take:b -->;`, or `take:d -->;` could be used to specify breadth first or depth first traversals respectively.

8.2 Ignoring and Deleting

Jac Code 8.4: Ignoring edges during walk

```
1 node person: has name;
2 edge family;
3 edge friend;
4
5 walker build_example {
6   spawn here -[friend]-> node::person(name="Joe");
7   spawn here -[friend]-> node::person(name="Susan");
8   spawn here -[family]-> node::person(name="Matt");
9   spawn here -[family]-> node::person(name="Dan");
10 }
11
12 walker init {
13   root {
14     spawn here walker::build_example;
15     ignore -[family]->;
16     ignore -[friend(name=="Joe")]->;
17     take -->;
18   }
19   person {
20     std.out(here.name);
21   }
22 }
```

Jac Code 8.5: Destroying nodes/edges during walk

```
1 node person: has name;
2 edge family;
3 edge friend;
4
```

```

5  walker build_example {
6      spawn here -[friend]-> node::person(name="Joe");
7      spawn here -[friend]-> node::person(name="Susan");
8      spawn here -[family]-> node::person(name="Matt");
9      spawn here -[family]-> node::person(name="Dan");
10 }
11
12 walker init {
13     root {
14         spawn here walker::build_example;
15         for i in -[friend]->: destroy i;
16         take -->;
17     }
18     person {
19         std.out(here.name);
20     }
21 }

```

8.3 Reporting Back as you Travel

Jac Code 8.6: Building reports as you walk

```

1  node person: has name;
2  edge family;
3  edge friend;
4
5  walker build_example {
6      spawn here -[friend]-> node::person(name="Joe");
7      spawn here -[friend]-> node::person(name="Susan");
8      spawn here -[family]-> node::person(name="Matt");
9      spawn here -[family]-> node::person(name="Dan");
10 }
11
12 walker init {
13     root {
14         spawn here walker::build_example;
15         spawn -->[0] walker::build_example;
16         take -->;
17     }
18     person {
19         report here; # report print back on disengage
20         take -->;
21     }
22 }

```

8.4 Yielding Walkers

So far, we've looked at walkers that will walk the graph carrying state in context (`has` variables). But you may wonder what happens after its walk? And does it keep that state like nodes and edges? Short answer is no. At the end of each walk a walker's state is cleared by default while node/edge state persists. That being said, there are situations where you'd want a walker to keep its state across runs, and perhaps, you may even want a walker to stop during a walk and wait to be explicitly called again updating just a few of its dynamic state. This is where the `yield` keyword comes in.

Lets look at an example of yield in action.

Jac Code 8.7: Simple example of yielding walkers

```

1  global node_count=0;
2
3  node simple {has id;}
4
5  walker simple_yield {
6      with entry {
7          t=here;
8          for i=0 to i<10 by i+=1 {
9              t = spawn t --> node::simple(id=global.node_count);
10             global.node_count+=1;
11         }
12     }
13     report here.context;
14     take -->;
15     yield;
16 }
```

The `yield` keyword in JC 8.7 instructs the walker `simple_yield` to stop walking and wait to be called again, even though the walker is instructed to `take -->` edges. In this example, a single next node location is queued up and the walker reports a single `here.context` each time it's called, taking only 1 edge per call.

8.4.1 Yield Shorthands

Also note `yield` can be followed by a number of operations as a shorthand. For example line 14 and 15 in JC 8.7 could be combined to a single line with `yield take -->;`. We call this a yield-take. Shorthands include,

- Yield-Take: `yield take -->;`
- Yield-Report: `yield report "hi";`

- Yield-Disengage: `yield disengage;` and `yield disengage report "bye";`

In each of these cases, the `take`, `report`, and `disengage` executes with the yield.

8.4.2 Technical Semantics of Yield

There are a number of important semantics of `yield` to keep in mind:

1. Upon a `yield`, a report is returned back and cleared.
2. Additional report items from further walking will be return on subsequent `yields` or walk completion.
3. Like the `take` command, the entire body of the walker will execute on the current node and actually yield at the end of this execution.
 - *Note: Keep in mind `yield` can be combined with `disengage` and `skip` commands.*
4. If a start node (aka a ‘prime’ node) is specified when continuing a walker after a `yield`, if there are additional walk locations the walker is scheduled to travel to, the walker will ignore this prime node and continue from where it left off on its journey.
5. If there are no nodes scheduled for the walker to go to next, a prime node must be specified (or the walker will continue from root by default).
6. `with entry` and `with exit` code blocks in the walker are not executed upon continuing from a `yield` or executing a `yeild` respectively. They execute only once starting and ending a walk though there may be many yields in between.
7. The state of which walkers are yielded and to be continued vs which walkers are being freshly run is kept at the level of the `master` (user) abstraction in Jaseci. At the moment, walkers that are summoned as public has undefined yield semantics. Developers should leverage the more lower level `walker spawn` and `walker execute` APIs for customized yield behaviors.

8.4.3 Walkers Yielding Other Walkers (i.e., Yielding Deeply)

In addition to the utility of calling walkers that yield from client, walkers also benefit from this abstraction when calling other walkers during a non-yielding walk. Lets take a look at a code example.

Jac Code 8.8: Walkers yielding other walkers

```

1  walker simple_yield {
2      with entry {
3          t=here;
4          for i=0 to i<4 by i+=1:
5              t = spawn t --> node::generic;
6          }
7      if(-->.length): yield take -->;
8  }
9
10 walker deep_yield {
11     for i=0 to i<16 by i+=1 {
12         spawn here walker::simple_yield;
13     }
14 }

```

As shown in JC 8.8, the walker `deep_yield` does not yield itself, but enjoys the semantics of the `yield` command in `simple_yield`.

Figure 8.2 shows the graph created by JC 8.8. Though `deep_yield` does not yield, it calls `simple_yield` 16 times and exits. These 16 calls trigger `walker::simple_yield` which in turn creates four chained nodes off of the root node then walks the chain one step at a time while yielding after each step. The result is this very nice 17 node graph with a root node and 3 subtrees with 4 connected nodes each. Yep, this yielding semantic is very handy indeed!




Figure 8.2: Graph in memory for JC 8.8

Chapter 9

Actions and Action Sets

9.1 Standard Action Library

9.1.1 date

Action	Args	Description
<code>date.quantize_to_year</code>	(date: str)	
<code>date.quantize_to_month</code>	(date: str)	
<code>date.quantize_to_week</code>	(date: str)	
<code>date.quantize_to_day</code>	(date: str)	
<code>date.date_day_diff</code>	(start_date: str, end_date: str)	

Table 9.1: Date Actions in Jac

9.1.2 file

Action	Args	Description
<code>file.load_str</code>	(fn: str, max_chars: int = None)	
<code>file.load_json</code>	(fn: str)	
<code>file.dump_str</code>	(fn: str, s: str)	
<code>file.append_str</code>	(fn: str, s: str)	
<code>file.dump_json</code>	(fn: str, obj, indent: int = None)	
<code>file.delete</code>	(fn: str)	

Table 9.2: Date Actions in Jac

9.2 Building Your Own Library

Chapter 10

Imports, File I/O, Tests, and More

10.1 Tests in Jac

Jac Code 10.1: Tests Example

```
1  node testnode {
2      has yo, bro;
3  }
4
5  node apple {
6      has v1, v2;
7  }
8
9  node banana {
10     has x1, x2;
11 }
12
13 graph dummy {
14     has anchor graph_root;
15     spawn {
16         graph_root = spawn node::testnode (yo="Hey_yo!");
17         n1=spawn node::apple(v1="I'm_apple");
18         n2=spawn node::banana(x1="I'm_banana");
19         graph_root --> n1 --> n2;
20     }
21 }
22
23 walker init {
24     has num=4;
25     report here.context;
```

```

26     report num;
27     take -->;
28 }
29
30 test "assert should be valid"
31 with graph::dummy by walker::init {
32     assert (num==4);
33     assert (here.x1=="I'm_banana");
34     assert <--[0].v1=="I'm_apple";
35 }
36
37 test "assert should fail"
38 with graph::dummy by walker::init {
39     assert (num==4);
40     assert (here.x1=="I'm_banana");
41     assert <--[0].v1=="I'm_Apple";
42 }
43
44 test "assert should fail, add internal except"
45 with graph::dummy by walker::init {
46     assert (num==4);
47     assert (here.x1=="I'm_banana");
48     assert <--[10].v1=="I'm_apple";
49 }

```

```

Testing "assert should be valid": [PASSED in 0.00s]
Testing "assert should fail": [FAILED in 0.00s]
('JAC Assert Failed', '<-- [ 0 ] . v1 == "I\'m Apple" ')
Testing "assert should fail, add internal except": [FAILED in 0.00s]
('JAC Assert Failed', '<-- [ 10 ] . v1 == "I\'m apple" ', IndexError('list index out of
↪ range'))
{
  "tests": 3,
  "passed": 1,
  "failed": 2,
  "success": false
}

```

10.2 Imports

Jac Code 10.2: Imports Example

```

1 import {graph::dummy, node::{banana, apple, testnode}} with "./jac_tests.jac";
2 # import {*} with "./jac_tests.jac";
3 # import {graph::dummy, node*} with "./jac_tests.jac";
4
5 walker init {
6     has num=4;
7     with entry {

```

```

8     spawn here --> graph::dummy;
9   }
10   report here.context;
11   report num;
12   take -->;
13 }

```

```

{
  "success": true,
  "report": [
    {},
    4,
    {
      "yo": "Hey yo!",
      "bro": null
    },
    4,
    {
      "x1": "I'm banana",
      "x2": null
    },
    4
  ]
}

```

10.3 File I/O

Jac Code 10.3: File I/O Example

```

1  walker init {
2    fn="fileiotest.txt";
3    a = {'a': 5};
4    file.dump_json(fn, a);
5    b=file.load_json(fn);
6    b['a']+=b['a'];
7    file.dump_json(fn, b);
8    c=file.load_str(fn);
9    file.append_str(fn, c);
10   c=file.load_str(fn);
11   report c;
12 }

```

10.4 Visualizing Graph with Dot Output

A very useful feature of the Jaseci stack is the ability to dump a snapshot of a graph in memory as `dot` output. There are two core interfaces to access this feature. The first is the `graph get` api. Simply set the `mode` parameter to “dot” and a dot representation of the graph will be printed. This API is present in both `jsctl` and the REST api. The other is to use `textttjac dot [filename]`. This will run the program specified in `filename`, then print the state of the graph at the end of the program run as dot output. This `jac dot` api is only available through `jsctl`. For both of these apis, a `detailed` parameter can be used to get more information embedded in the dot output. In particular, any context variables that are string will be included in the nodes and edges of the dot output.

Part III

Jaseci AI Kit

Part IV

Crafting Jaseci

Chapter 11

Architecting Jaseci Core

Chapter 12

Architecting Jaseci Cloud Serving

Epilogue

Appendix A

Jumping Right In, TLDR style

If you're the kind of haxor that doesn't want to read a huge book and just wants to get hacking ASAP, this chapter is for you!! This chapter will make a few assumptions. Firstly, it is assumed that you are in a linux environment and will have command of the line that takes commands. Coincidental, this is commonly referred to as the *command line*. Secondly, this command line will be one that accepts linux style commands in a **bash** format. If you've never heard of bash, Google it. Thirdly and lastly, you will be using the only IDE true ninjas use, namely **VSCode**. If these conditions apply to your environment, you're good. If they don't but you use Linux, you're still good (as you're almost certainly competent enough at this stuff to be able to easily be able to make the necessary adjustments to get things working in your environment.)

We start this journey from the perspective of having a fresh vanilla install of the minimal version of Ubuntu 20+. Ubuntu is a distribution or (flavor) of linux that is likely the most popular and accessible in the market. I say likely because I don't know for sure, but if it isn't I'd be shocked!

A.1 Installation

First and foremost, lets check what os we're running at the moment.

```
haxor@linux:~$ cat /etc/os-release
NAME="Ubuntu"
VERSION="20.04.4 LTS (Focal Fossa)"
ID=ubuntu
ID_LIKE=debian
PRETTY_NAME="Ubuntu 20.04.4 LTS"
VERSION_ID="20.04"
HOME_URL="https://www.ubuntu.com/"
SUPPORT_URL="https://help.ubuntu.com/"
BUG_REPORT_URL="https://bugs.launchpad.net/ubuntu/"
PRIVACY_POLICY_URL="https://www.ubuntu.com/legal/terms-and-policies/privacy-policy"
VERSION_CODENAME=focal
UBUNTU_CODENAME=focal
haxor@linux:~$
```

Ok good, we're running Ubuntu 20.04.4 LTS as the `PRETTY_NAME=` indicates.

Now immediately execute `sudo apt update` and `sudo apt upgrade` as two separate commands, don't ask why just do it.

A.1.1 Python Environment

Next, we need to have Python installed. Python is the programming language and runtime that Jaseci is primarily built upon. It's also the language that 99.999% of everyone uses for AI research and products (and myriad other things). It's also my favorite as of late, well, second favorite after Jac. Lets check to see. Simply enter the command,

```
haxor@linux:~$ python3 --version
-bash: python3: command not found
haxor@linux:~$
```

Some of you at this point might see a python version that is ≥ 3.8 . If you see this you're good, you have Python installed. We don't see this in this example. That is because we have the *minimal* Ubuntu. So we have to install it.

```

haxor@linux:~$ sudo apt install python3 python3-pip
Reading package lists... Done
Building dependency tree
Reading state information... Done
The following additional packages will be installed:
  binutils binutils-common binutils-x86-64-linux-gnu build-essential ca-certificates cpp
  ↳ cpp-9 dirmngr dpkg-dev fakeroot g++ g++-9 gcc gcc-9 gcc-9-base gnupg
...
Do you want to continue? [Y/n] y
...
Processing triggers for libc-bin (2.31-0ubuntu9.7) ...
Processing triggers for ca-certificates (20210119~20.04.2) ...
Updating certificates in /etc/ssl/certs...
0 added, 0 removed; done.
Running hooks in /etc/ca-certificates/update.d...
done.
haxor@linux:~$

```

The line `sudo apt install python3 python3-pip` instructs Ubuntu to install both the `python3` package as well as the `python3-pip` package. Note in the example there is a point where it will ask you if you want to continue, just press Y and let it go. This step could take some time in principle, but we are almost there!

Lets next check again that we have python installed.

```

haxor@linux:~$ python3 --version
Python 3.8.10
haxor@linux:~$ pip --version
pip 20.0.2 from /usr/lib/python3/dist-packages/pip (python 3.8)
haxor@linux:~$

```

Yes! We're in great shape, we've also checked that `pip` is install and that looks good as well. Note that we can also replace `pip` with `pip3` and everything should work as well.

A.1.2 Installing Jaseci

Now that we have Python setup, we can use the `pip` install Jaseci itself. `pip` is Python's official package manager. This command line tool allows users of Python to install packages or code libraries that go beyond the standard libraries that come with Python out of the box. There is a public repository of libraries that is open to all the haxors of the world called PyPI [7] that houses pretty much all the published python packages of the world. Jaseci lives there throuh two packages, `jaseci` and `jaseci-serv`. For the moment we need only concern ourselves with `jaseci` as we get started. When we're ready to launch amazing tech stacks to production on scalable cloud infrastructure we'll pull down `jaseci-serv`.

Now, lets install Jaseci!

```
haxor@linux:~$ pip install jaseci
Collecting jaseci
  Downloading jaseci-1.3.1.1-py3-none-any.whl (154 kB)
    | 154 kB 4.5 MB/s
...
Successfully installed jaseci-1.3.1.1
haxor@linux:~$
```

TADA! We've pulled down Jaseci and are good to go! In this case we've installed Jaseci version 1.3.1.1, your version should be at least this one but probably higher depending on when you're reading this. If its say a year after this moment that I'm writing this book and it's still 1.3.1.1, something very very wrong has happened. Indeed, if its two weeks later and nothing has changed, call 911 and report a missing person, seriously.

To validate that everything works, lets check the command line tool `jsctl` is present. `jsctl` is a command line tool that give full control and access to the Jaseci computational model. In particular, and for the sake of this chapter, we will use this tool to build and run programs, generate source for visualizing data and graphs, building artificial intelligence (AI) programs, hot loading fancy AI models, pushing implementations live to Jaseci servers and much much more. Now lets make sure we have access to this very powerful cli tool.

```

haxor@linux:~$ jsctl --help
Usage: jsctl [OPTIONS] COMMAND [ARGS]...

The Jaseci Command Line Interface

Options:
  -f, --filename TEXT Specify filename for session state.
  -m, --mem-only Set true to not save file for session.
  --help Show this message and exit.

Commands:
  actions Group of 'actions' commands
  alias Group of 'alias' commands
  archetype Group of 'architype' commands
  clear Clear terminal
  config Group of 'config' commands
  edit Edit a file
  global Group of 'global' commands
  graph Group of 'graph' commands
  jac Group of 'jac' commands
  logger Group of 'logger' commands
  login Command to log into live Jaseci server
  ls List relevant files
  master Group of 'master' commands
  object Group of 'object' commands
  reset Reset jsctl (clears state)
  sentinel Group of 'sentinel' commands
  stripe Group of 'stripe' commands
  tool Internal book generation tools
  walker Group of 'walker' commands
haxor@linux:~$

```

If you see this output, you're in business!! If you don't, something went wrong and you should phone a friend, (but first make sure you didn't miss anything above).

A.1.3 VSCode and the Jac Language Extension

This is technically optional but... I strongly recommend you install and use VSCode with Jaseci. VSCode IMHO, is the best code editor on the planet. I regard it as the choice Sake to sip with my Jaseci omakase. Personally, I use an Ubuntu flavored WSL VSCode environment.

In VSCode, you can search for and install the Jac language extensions as per Figure A.1. As you can see, at the time I clipped this image, its quite new and doesn't really have a readme. You won't need one, it just provides syntax highlighting for `.jac` files at the moment. But it makes Jac code look beautiful, so it's a must have.




Figure A.1: The Wonderful Jac Language extension in VSCode.

A.2 Coding in Jac

Jac, which is short hand for **Jaseci Code**, is a programming language designed for building programs for Jaseci. The language itself is inspired by a mixture of Javascript and Python and can be used standalone or as glue code for libraries built in other languages ecosystems. Jac is to Python, what Python is to C, what C is to assembly language for scalable sophisticated applications running in the cloud. In this section, we'll cover basics to advanced assuming no programming experience. Though we'll try to cover everything from first time coders to pros, we'll move fast through some of the rudimentary concepts so have your Google ready if you need to drill in a bit more of some of the basic programming concepts. Lets Jump in!

A.2.1 Jac Basics

Launch VSCode, spool up a terminal window, and lets tinker with an example. We'll start with Jac Code A.1. I'd strongly recommend you type out this example (instead of cutting and pasting) especially if this might be your first time programming or are a little rusty with Python and or Javascript. It's the best way to learn!

```
Jac Code A.1: Example program introducing basic syntax.
1 walker init {
```

```

2  x = 34 - 30; # This is a comment
3  y = "Hello";
4  z = 3.45;
5
6  if(z==3.45 or y=="Bye"){ # if statement with only thing true
7      x=x-1;
8      y=y+" World"; # the + between two strings concatenate them
9  }
10
11  std.out(x);
12  for i=0 to i<3 by i+=1: # For loop with single line block style
13      std.out(x-i, '-', y); # prints to screen
14      report [x, y+'s']; # adds data to payload
15  }

```

This first example Jac Code A.1 shows a simple program example demonstrating a number of basic language features. Firstly, observe that the first three assignments in the program to `x`, `y`, and `z` does not specify any types indicating that Jac is a dynamically typed language. This means the types are inferred from the assignment of variables, and these types can change dynamically as new assignments are applied to the same variables. This feature is designed to work almost exactly like the dynamic typing in Python.

Next we find a conditional statement much like any other language. Do note operators like the Python inspired `or` is supported along side the C/C++/Javascript `||` operator. Other such operators include `and` (`\&\&`), `not` (`!`), etc.

After the conditional we have a library call `std.out(x)` on line 11. This call prints the value of `x` to the screen. `std.out` in Jac is equivalent to the `print` in Python and analogous to the `printf`, `cout`, and `console.log` you'd find in C, C++, and Javascript respectively. A suite of core standard library operations for the language has the preamble of `std`.

Output:

```

3
3 - Hello World
2 - Hello World
1 - Hello World
{
  "success": true,
  "report": [
    [
      3,
      "Hello Worlds"
    ]
  ]
}

```

A.2.2 Types in Jac

[Types example]

```
Jac Code A.2: First Example
1  walker init {
2      a=5;
3      b=5.0;
4      c=true;
5      d='5';
6      e=[a, b, c, d, 5];
7      f={'num': 5};
8
9      summary = {'int': a, 'float': b, 'bool': c,
10                'string': d, 'list': e, 'dict': f};
11
12      std.out(summary);
13 }
```

Output:

```
{"int": 5, "float": 5.0, "bool": true, "string": "5", "list": [5, 5.0, true, "5", 5], "
↪ dict": {"num": 5}}
```

A.2.3 Fun with Lists and Dictionaries

[Fun with Lists and Dictionaries]

```
Jac Code A.3: First Example
1  walker init {
2      d = {'four':4, 'five':5};
3      b = d.dict::copy; # equal to b=d.d::copy;
4      b['four'] += b['five'];
5      std.out(d.d::keys, d.d::values, d.d::items, b.d::items);
6
7      b_vals = b.d::values;
8      b_vals.list::append(6.5); # equal to b=d.d::copy;
9      std.out(b_vals);
10     b_vals.l::sort; std.out(b_vals);
11     b_vals.l::reverse; std.out(b_vals);
12 }
```

Output:

```
["four", "five"] [4, 5] [{"four", 4}, {"five", 5}] [{"four", 9}, {"five", 5}]
[9, 5, 6.5]
[5, 6.5, 9]
[9, 6.5, 5]
```

A.2.4 Control Flow

[Fun with Control Flow]

Jac Code A.4: First Example

```

1  walker init {
2      fav_nums=[];
3
4      for i=0 to i<10 by i+=1:
5          fav_nums.1::append(i*2);
6          std.out(fav_nums);
7
8      fancy_str = "";
9      for i in fav_nums {
10         fancy_str = fancy_str + "two_␣" + i.str +
11             "␣" + (i*2).str + ",␣";
12     }
13     std.out(fancy_str);
14
15     count_down = fav_nums[-1];
16     while (count_down > 0) {
17         count_down -= 1;
18         if (count_down == 14):
19             continue;
20         std.out("I'm at_␣countdown_␣"+count_down.str);
21         if (count_down == 10):
22             break;
23     }
24 }
```

Output:

```

[0, 2, 4, 6, 8, 10, 12, 14, 16, 18]
two * 0 = 0, two * 2 = 4, two * 4 = 8, two * 6 = 12, two * 8 = 16, two * 10 = 20, two * 12
↪ = 24, two * 14 = 28, two * 16 = 32, two * 18 = 36,
I'm at countdown 17
I'm at countdown 16
I'm at countdown 15
I'm at countdown 13
I'm at countdown 12
I'm at countdown 11
I'm at countdown 10
```




Figure A.2: Graph in memory for JC A.5

A.2.5 Graphs in Jac

[Bringing Graphs in with special operators]

Jac Code A.5: First Example

```

1  node person {
2      has name="Anon";
3  }
4
5  edge strong;
6  edge weak;
7
8  walker init {
9      person1 = spawn here -[strong]-> node::person(name="Joe");
10     person2 = spawn here -[strong]-> node::person;
11     person3 = spawn here -[weak]-> node::person;
12     person4 = spawn here -[weak]-> node::person(name="Mike");
13
14     person1 <--> person2;
15     person3 <-[strong]-> person4;
16
17     for i in -->:
18         std.out(i.context);
19 }

```

Output:

```

{"name": "Joe"}
{"name": "Anon"}
{"name": "Anon"}
{"name": "Mike"}

```

A.2.6 Navigating Graphs with Walkers

[Walking Graphs]

Jac Code A.6: First Example

```

1  node state {
2      has response="I'm_silly_state_";
3  }
4
5  node hop_state;
6
7  edge hop;
8
9  walker init {
10     has state_visits=0, save_root;
11
12     root {
13         save_root = here;
14         hop1 = spawn here -[hop]-> node::hop_state;
15         hop2 = spawn here -[hop]-> node::hop_state;
16     }
17
18     hop_state:
19         spawn here walker::hop_buildout;
20
21     state {
22         state_visits += 1;
23         std.out(here.response+state_visits.str);
24     }
25
26     take -->;
27     with exit {
28         report spawn save_root walker::hop_counter;
29     }
30 }
31
32 walker hop_buildout {
33     spawn here --> node::state;
34     spawn here --> node::state;
35     spawn here --> node::state;
36 }
37
38 walker hop_counter {
39     has anchor num=0; take -->; hop_state { num+=1; }
40 }

```





Figure A.3: Graph in memory for JC A.6

Output:

```

I'm silly state 1
I'm silly state 2
I'm silly state 3
I'm silly state 4
I'm silly state 5
I'm silly state 6
{
  "success": true,
  "report": [
    2
  ]
}

```

A.2.7 Compute in Nodes

[Compute into the Nodes]

Jac Code A.7: First Example

```

1 node state {
2   has name = rand.word().str::upper;
3   has response = "I'm a silly bot.";
4   has user_utter;
5
6   can speak with entry {
7     std.out("I'm " + name + ". And I currently have " + visitor.info['name'] +
8       " on me!");
9   }
10
11   can listen with talker exit {
12     user_utter = visitor.utterance;
13     std.out("I heard " + user_utter + "\n");
14     std.out(response);
  
```




Figure A.4: Graph in memory for JC A.7

```

15 }
16
17 can test_path with hop_counter entry {
18     visitor.path.1::append(&here);
19 }
20 }
21
22 walker init {
23     root {
24         n1 = spawn here --> node::state;
25         n2 = spawn here --> node::state;
26     }
27     spawn here walker::talker;
28     spawn here walker::hop_counter;
29 }
30
31 walker talker {
32     has utterance, path = [];
33     utterance = rand.sentence();
34     take -->;
35 }
36
37 walker hop_counter {
38     has anchor path = [];
39     take -->;
40
41     with exit { std.out("\nHopper's_path:", path); }
42 }

```

Output:

```

I'm DOLOREM. And I currently have talker on me!
I heard 'Magnam quaerat ut qui velit consectetur consectetur.'

I'm a silly bot.
I'm EIUS. And I currently have talker on me!
I heard 'Quisquam eius numquam amet ut porro velit amet numquam ut.'

I'm a silly bot.
I'm DOLOREM. And I currently have hop_counter on me!
I'm EIUS. And I currently have hop_counter on me!

Hopper's path: ["urn:uuid:d5be01eb-db6f-4692-9471-05ccf081ffc1", "urn:uuid:e7dd97bf-050c-4
↳ b36-afa5-38963935c933"]

```

A.2.8 Static Graphs

[Static graphs]

Jac Code A.8: First Example

```

1  node person {
2      has name="Anon";
3  }
4
5  edge strong;
6  edge weak;
7
8  graph basic_gph {
9      has anchor root;
10     spawn {
11         root = spawn node::generic;
12         person1 = spawn root -[strong]-> node::person(name="Joe");
13         person2 = spawn root -[strong]-> node::person;
14         person3 = spawn root -[weak]-> node::person;
15         person4 = spawn root -[weak]-> node::person(name="Mike");
16
17         person1 <--> person2;
18         person3 <-[strong]-> person4;
19     }
20 }
21
22
23 walker init {
24     spawn here --> graph::basic_gph;
25     spawn here --> graph::basic_gph;
26     spawn here --> graph::basic_gph;
27 }

```




Figure A.5: Graph in memory for JC A.8

A.2.9 Writing Tests

[Tests]

Jac Code A.9: First Example

```

1 node person: has name="Anon";
2
3 graph basic {
4   has anchor root;
5   spawn {
6     root = spawn node::generic;
7     person1 = spawn root --> node::person(name="Joe");
8     person2 = spawn root --> node::person;
9     person3 = spawn root --> node::person;
10    person4 = spawn root --> node::person(name="Mike");
11    person1 <--> person2;
12    person3 <--> person4;
13  }
14 }
15
16
17 walker tally {
18   has count=0, visited=[];
19   count += 1;
20
21   if(here not in visited) {
22     visited.l::append(here);
23     take -->;
24   }
25 }
26
27 test "Size_of_basic_graph"
28 with graph::basic by walker::tally {
29   assert(visited.length == 5);
30   assert(count > 5);
31 }
32

```

```

33 test "Size of a bit fancier graph"
34 with graph {
35   has anchor root;
36   spawn {
37     root = spawn node::generic;
38     spawn root --> graph::basic; spawn root --> graph::basic;
39   }
40 } by walker::tally {
41   assert(visited.length == 11);
42   assert(count > 11);
43 }

```

Output:

```

Testing "Size of basic graph": [PASSED in 0.00s]
Testing "Size of a bit fancier graph": [PASSED in 0.01s]
{
  "tests": 2,
  "passed": 2,
  "failed": 0,
  "success": true
}

```

A.3 Jac Hacking Workflow

In this section, we discuss a typical workflow and organization of a Jac coding project. To this end, we will be creating a simple toy chatbot project and examine it's file organization and development workflow. First, lets take a look at the files for this project.

```

haxor@linux:~/toybot$ ls
cai.jac edges.jac faq_answers.txt load_faq.jac nodes.jac static_conv.jac tests.jac
haxor@linux:~$

```

Now lets take a look a what each of these files represent:

- **cai.jac** - This is the main file for the project to which the various other elements (nodes, edges, graphs, etc) are imported from other files in the directory.
- **nodes.jac** - This file houses the node architypes created for this application. Functionality is specified in both the walkers and as node abilities.
- **edges.jac** - This file contains the edge architypes we've specified in the design of our conversational AI. These edges represent various types of transitions we can make throughout the converstation.

- `static_conv.jac` - This file contains a static conversational graph that represents the possible conversational flows via state nodes and transition edges.
- `load_faq.jac` - This file contains a static constructor for graph elements to correspond to frequently asked questions by loading them from a file.
- `faq_answers.txt` - This file specifies a list of answers to frequently asked questions, we'll be using a model that only depends on the answers themselves.
- `tests.jac` - This file is where we house all the tests for our project.

A.3.1 Using Imports

Jac Code A.10: Main CAI Jac App

```

1 import {node::{state, hop_state}} with "./nodes.jac";
2 import {edge::{trans_ner, trans_intent, trans_qa}} with "./edges.jac";
3 import {graph::basic_gph} with "./static_conv.jac";
4 import {graph::faq_gph} with "./load_faq.jac";
5
6
7
8 walker init {
9   root {
10     spawn here --> graph::basic_gph;
11     spawn -->[0] -[trans_intent(intent="about_chat_bots")]-> graph::faq_gph;
12   }
13   with exit {
14     spawn -->[0] walker::talker;
15   }
16 }
17
18 walker talker {
19   has utterance="";
20   has use_cmd = true, path = [];
21   if(use_cmd and here.details['name'] != 'hop_state'):
22     utterance = std.input(">");
23   take -->;
24 }

```

Jac Code A.11: Nodes for CAI

```

1 node state {
2   has name = rand.word();
3   has response="I'm a silly bot.";
4   has user_utter;
5
6   can speak with entry {
7     std.out(response + " I'm current on " + name + " node");
8   }
9 }

```

```

10     can listen with talker exit {
11         user_utter = visitor.utterance;
12         visitor.path.1::append(&here);
13         std.out("I_heard_" + user_utter + ".");
14     }
15
16     can test_path with get_states entry {
17         visitor.path.1::append(&here);
18     }
19 }
20
21 node hop_state {
22     has name;
23     can log with exit {
24         std.log("A_walker_is_walking_right_over_me.");
25     }
26 }

```

Jac Code A.12: edges for CAI

```
1 edge trans_ner { has entities; }
2 edge trans_intent { has intent; }
3 edge trans_qa { has embed; }
```

A.3.2 Leveraging Static Graphs for Quick Prototyping

Jac Code A.13: Static Conversational Graph

```

1 import {edge::{trans_ner, trans_intent, trans_qa}} with "./edges.jac";
2 import {node::{state, hop_state}} with "./nodes.jac";
3
4 graph basic_gph {
5     has anchor conv_root;
6     spawn {
7         conv_root = spawn node::state(name="Conv_Root");
8
9         appt = spawn conv_root -[trans_intent(intent="appointment")]->
10             node::hop_state(name="Appointments");
11
12         spawn appt -[trans_intent(intent="create")]->
13             node::state(name="Create_an_appointment");
14         spawn appt -[trans_intent(intent="cancel")]->
15             node::state(name="Cancel_an_appointment");
16         spawn appt -[trans_intent(intent="reschedule")]->
17             node::state(name="Reschedule_an_appointment");
18
19         service = spawn conv_root -[trans_intent(intent="service_info")]->
20             node::hop_state(name="Services");
21
22         spawn service -[trans_intent(intent="manicures")]->
23             node::state(name="About_manicures");

```

```

24     spawn service -[trans_intent(intent="haircuts")]->
25         node::state(name="About_haircuts");
26     spawn service -[trans_intent(intent="makeup")]->
27         node::state(name="About_makeup");
28 }
29
30 }

```

A.3.3 Test Driven Development

Jac Code A.14: Tests for CAI

```

1  import {*} with "./cai.jac";
2
3  walker get_states {
4      has anchor path = [];
5      take -->;
6  }
7
8  test "Travesal_touches_all_nodes"
9  with graph::basic_gph by walker::get_states {
10     std.out(path.length);
11     assert(path.length==7);
12 }

```

A.3.4 File I/O

Jac Code A.15: FAQ Graph Loader

```

1  import {edge::{trans_ner, trans_intent, trans_qa}} with "./edges.jac";
2  import {node::{state, hop_state}} with "./nodes.jac";
3
4  graph faq_gph {
5      has anchor faq_root;
6      spawn {
7          faq_root = spawn node::state(name="FaQ_Root");
8
9          answers = file.load_str('./faq_answers.txt').str::split('&&&');
10
11          for i in answers:
12              spawn faq_root -[trans_qa]-> node::state(response=i);
13      }
14
15 }

```



```

A chatbot is an artificial intelligence (AI) based computer program that can interact with
    ↪ a human either via voice or text through messaging applications, websites, mobile
    ↪ apps or through the telephone.
&&&
Conversational chatbots have been around for decades now. In the past, there have been
    ↪ many unsuccessful attempts to build a chatbot that successfully mimics human
    ↪ conversation. However, not thats solved with the creation of me!
&&&
During the chatbot design process, it is important to keep your user in mind as it will
    ↪ help you define the right chatbot features, functionality and build human-like
    ↪ interactions.
&&&
In order for a chatbot to function properly, it is crucial for the program to access your
    ↪ knowledge base, website, internal databases, existing documents, or other sources
    ↪ of information.

```

A.3.5 Building to JIR

A.4 AI with Jaseci Kit

A.4.1 Installing Jaseci Kit

```

haxor@linux:~$ pip install jaseci-kit
Collecting jaseci-kit
  Downloading jaseci_kit-1.3.3.5-py3-none-any.whl (34 kB)
Collecting tensorflow<3.0.0,>=2.8.0
  Downloading tensorflow-2.8.0-cp38-cp38-manylinux2010_x86_64.whl (497.6 MB)
    | 497.6 MB 8.9 MB/s
...
Successfully installed ... jaseci-kit-1.3.3.5 ...
haxor@linux:~$

```

A.4.2 Loading Actions from Jaseci Kit

```

haxor@linux:~$ jsctl -m
Starting Jaseci Shell...
jaseci > actions list
[
  "net.max",
  "net.min",
  "net.root",
  "rand.seed",
  ...
  "date.quantize_to_month",
  "date.quantize_to_week",
  "date.quantize_to_day",
  "date.date_day_diff"
]
jaseci >

```

```

jaseci > actions load module jaseci_kit.use_qa
2022-04-16 22:01:52.612881: W tensorflow/stream_executor/platform/default/dso_loader.cc
  ↳ :64] Could not load dynamic library 'libcudart.so.11.0'; dlerror: libcudart.so
  ↳ .11.0: cannot open shared object file: No such file or directory
2022-04-16 22:01:52.612908: I tensorflow/stream_executor/cuda/cudart_stub.cc:29] Ignore
  ↳ above cudart dlerror if you do not have a GPU set up on your machine.
2022-04-16 22:02:05.269074: W tensorflow/stream_executor/platform/default/dso_loader.cc
  ↳ :64] Could not load dynamic library 'libcuda.so.1'; dlerror: libcuda.so.1: cannot
  ↳ open shared object file: No such file or directory
2022-04-16 22:02:05.269104: W tensorflow/stream_executor/cuda/cuda_driver.cc:269] failed
  ↳ call to cuInit: UNKNOWN ERROR (303)
2022-04-16 22:02:05.269127: I tensorflow/stream_executor/cuda/cuda_diagnostics.cc:156]
  ↳ kernel driver does not appear to be running on this host (vanillabox-589f9b897c-
  ↳ k2ncs): /proc/driver/nvidia/version does not exist
2022-04-16 22:02:05.269232: I tensorflow/core/platform/cpu_feature_guard.cc:151] This
  ↳ TensorFlow binary is optimized with oneAPI Deep Neural Network Library (oneDNN) to
  ↳ use the following CPU instructions in performance-critical operations: AVX2 FMA
To enable them in other operations, rebuild TensorFlow with the appropriate compiler flags
  ↳ .
{
  "success": true
}
jaseci >

```

```
jaseci > actions list
[
  "net.max",
  "net.min",
  "net.root",
  "rand.seed",
  ...
  "date.quantize_to_month",
  "date.quantize_to_week",
  "date.quantize_to_day",
  "date.date_day_diff",
  "use.question_encode",
  "use.enc_question",
  "use.answer_encode",
  "use.enc_answer",
  "use.cos_sim_score",
  "use.dist_score",
  "use.qa_score"
]
jaseci >
```

A.4.3 Using AI in Jac

[Adding some AI]

```
Jac Code A.16: Universal Sentence Encoding QA in Jac
1  walker init {
2    can use.enc_question, use.enc_answer;
3
4    answers = ['I_am_20_years_old', 'My_dog_is_hungry', 'My_TV_is_broken'];
5    question = "If_I_wanted_to_fix_something_what_should_I_fix?";
6
7    q_enc = use.enc_question(question);
8    a_enc = use.enc_answer(answers); # can take lists or single strings
9
10   a_scores=[];
11
12   for i in a_enc:
13     a_scores.1::append(vector.cosine_sim(q_enc, i));
14
15   report a_scores;
16 }
```

Output:

```
{  
  "success": true,  
  "report": [  
    0.010415227400767156,  
    0.034413563053388725,  
    0.08458081860660219  
  ]  
}
```

A.5 Launching a Jaseci Web Server

A.6 Deploying Jaseci at Scale

A.6.1 Quick-start with Kubectl

A.6.2 Managing Jac in Cloud

Appendix B

Rants

B.1 Utilizing Whitespace for Scoping is Criminal (Yea, I'm looking at you Python)

This whitespace debauchery perpetrated by Python and the like is one of the most perverse abuses of ASCII code 32 I've seen in computer science. It's an assault on the freedom of coders to decide the shape and structure of the beautiful sculptures their creative minds might want to actualize in syntax. Coder's fingers have a voice! And that voice deserves to be heard! The only folks that support this oppression are those in the 1% that get paid on a per line of code basis so they can lean on these whitespace mandates to pump up their salaries at the cost of coders everywhere.

“FREE THE PEOPLE! FREE THE CODE!”

“FREE THE PEOPLE! FREE THE CODE!”

“FREE THE PEOPLE! FREE THE CODE!”

B.2 “Using ‘master’ in Git is NOT Racist!” by a black dude

[Insert Rant Here]

Appendix C

Full Jac Grammar Specification

Grammar C.1: Full listing of Jac Grammar (antlr4)

```
1 grammar jac;
2
3 start: ver_label? element+ EOF;
4
5 element: architype | walker;
6
7 architype:
8     KW_NODE NAME (COLON INT)? attr_block
9     | KW_EDGE NAME attr_block
10    | KW_GRAPH NAME graph_block;
11
12 walker:
13     KW_WALKER NAME namespaces? LBRACE attr_stmt* walk_entry_block? (
14         statement
15         | walk_activity_block
16     )* walk_exit_block? RBACE;
17
18 ver_label: 'version' COLON STRING SEMI?;
19
20 namespaces: COLON name_list;
21
22 walk_entry_block: KW_WITH KW_ENTRY code_block;
23
24 walk_exit_block: KW_WITH KW_EXIT code_block;
25
26 walk_activity_block: KW_WITH KW_ACTIVITY code_block;
27
28 attr_block: LBRACE (attr_stmt)* RBACE | COLON attr_stmt | SEMI;
29
30 attr_stmt: has_stmt | can_stmt;
31
```

```

32 graph_block: graph_block_spawn | graph_block_dot;
33
34 graph_block_spawn:
35     LBRACE has_root KW_SPAWN code_block RBRACE
36     | COLON has_root KW_SPAWN code_block SEMI;
37
38 graph_block_dot:
39     LBRACE has_root dot_graph RBRACE
40     | COLON has_root dot_graph SEMI;
41
42 has_root: KW_HAS KW_ANCHOR NAME SEMI;
43
44 has_stmt:
45     KW_HAS KW_PRIVATE? KW_ANCHOR? has_assign (COMMA has_assign)* SEMI;
46
47 has_assign: NAME | NAME EQ expression;
48
49 can_stmt:
50     KW_CAN dotted_name (preset_in_out event_clause)? (
51         COMMA dotted_name (preset_in_out event_clause)?
52     )* SEMI
53     | KW_CAN NAME event_clause? code_block;
54
55 event_clause:
56     KW_WITH name_list? (KW_ENTRY | KW_EXIT | KW_ACTIVITY);
57
58 preset_in_out:
59     DBL_COLON expr_list? (DBL_COLON | COLON_OUT expression);
60
61 dotted_name: NAME DOT NAME;
62
63 name_list: NAME (COMMA NAME)*;
64
65 expr_list: expression (COMMA expression)*;
66
67 code_block: LBRACE statement* RBRACE | COLON statement;
68
69 node_ctx_block: name_list code_block;
70
71 statement:
72     code_block
73     | node_ctx_block
74     | expression SEMI
75     | if_stmt
76     | for_stmt
77     | while_stmt
78     | ctrl_stmt SEMI
79     | destroy_action
80     | report_action
81     | walker_action;
82
83 if_stmt: KW_IF expression code_block (elif_stmt)* (else_stmt)?;
84
85 elif_stmt: KW_ELIF expression code_block;
86

```

```

87 else_stmt: KW_ELSE code_block;
88
89 for_stmt:
90     KW_FOR expression KW_TO expression KW_BY expression code_block
91     | KW_FOR NAME KW_IN expression code_block;
92
93 while_stmt: KW_WHILE expression code_block;
94
95 ctrl_stmt: KW_CONTINUE | KW_BREAK | KW_SKIP;
96
97 destroy_action: KW_DESTROY expression SEMI;
98
99 report_action: KW_REPORT expression SEMI;
100
101 walker_action: ignore_action | take_action | KW_DISENGAGE SEMI;
102
103 ignore_action: KW_IGNORE expression SEMI;
104
105 take_action: KW_TAKE expression (SEMI | else_stmt);
106
107 expression: connect (assignment | copy_assign | inc_assign)?;
108
109 assignment: EQ expression;
110
111 copy_assign: CPY_EQ expression;
112
113 inc_assign: (PEQ | MEQ | TEQ | DEQ) expression;
114
115 connect: logical ( (NOT)? edge_ref expression)?;
116
117 logical: compare ((KW_AND | KW_OR) compare)*;
118
119 compare: NOT compare | arithmetic (cmp_op arithmetic)*;
120
121 cmp_op: EE | LT | GT | LTE | GTE | NE | KW_IN | nin;
122
123 nin: NOT KW_IN;
124
125 arithmetic: term ((PLUS | MINUS) term)*;
126
127 term: factor ((MUL | DIV | MOD) factor)*;
128
129 factor: (PLUS | MINUS) factor | power;
130
131 power: func_call (POW factor)*;
132
133 func_call:
134     atom (LPAREN expr_list? RPAREN)?
135     | atom? DBL_COLON NAME spawn_ctx?;
136
137 atom:
138     INT
139     | FLOAT
140     | STRING
141     | BOOL

```



```

142 | NULL
143 | NAME
144 | node_edge_ref
145 | list_val
146 | dict_val
147 | LPAREN expression RPAREN
148 | spawn
149 | atom DOT built_in
150 | atom DOT NAME
151 | atom index_slice
152 | ref
153 | deref
154 | any_type;
155
156 ref: '&' expression;
157
158 deref: '*' expression;
159
160 built_in:
161     cast_built_in
162     | obj_built_in
163     | dict_built_in
164     | list_built_in
165     | string_built_in;
166
167 cast_built_in: any_type;
168
169 obj_built_in: KW_CONTEXT | KW_INFO | KW_DETAILS;
170
171 dict_built_in: KW_KEYS | LBRACE name_list RBRACE;
172
173 list_built_in: KW_LENGTH | KW_DESTROY COLON expression COLON;
174
175 string_built_in:
176     TYP_STRING DBL_COLON NAME (LPAREN expr_list RPAREN)?;
177
178 node_edge_ref:
179     node_ref filter_ctx?
180     | edge_ref (node_ref filter_ctx)?;
181
182 node_ref: KW_NODE DBL_COLON NAME;
183
184 walker_ref: KW_WALKER DBL_COLON NAME;
185
186 graph_ref: KW_GRAPH DBL_COLON NAME;
187
188 edge_ref: edge_to | edge_from | edge_any;
189
190 edge_to:
191     '-->'
192     | '->' ('[' NAME (spawn_ctx | filter_ctx)? ']')? '->';
193
194 edge_from:
195     '<--'
196     | '<-' ('[' NAME (spawn_ctx | filter_ctx)? ']')? '<-';

```

```

197
198 edge_any:
199     '<-->'
200     | '<->' ('[' NAME (spawn_ctx | filter_ctx)? '']')? '->';
201
202 list_val: LSQUARE expr_list? RSQUARE;
203
204 index_slice:
205     LSQUARE expression RSQUARE
206     | LSQUARE expression COLON expression RSQUARE;
207
208 dict_val: LBRACE (kv_pair (COMMA kv_pair)*)? RBRACE;
209
210 kv_pair: STRING COLON expression;
211
212 spawn: KW_SPAWN expression? spawn_object;
213
214 spawn_object: node_spawn | walker_spawn | graph_spawn;
215
216 node_spawn: edge_ref? node_ref spawn_ctx?;
217
218 graph_spawn: edge_ref graph_ref;
219
220 walker_spawn: walker_ref spawn_ctx?;
221
222 spawn_ctx: LPAREN (spawn_assign (COMMA spawn_assign)*)? RPAREN;
223
224 filter_ctx:
225     LPAREN (filter_compare (COMMA filter_compare)*)? RPAREN;
226
227 spawn_assign: NAME EQ expression;
228
229 filter_compare: NAME cmp_op expression;
230
231 any_type:
232     TYP_STRING
233     | TYP_INT
234     | TYP_FLOAT
235     | TYP_LIST
236     | TYP_DICT
237     | TYP_BOOL
238     | KW_NODE
239     | KW_EDGE
240     | KW_TYPE;
241
242 /* DOT grammar below */
243 dot_graph:
244     KW_STRICT? (KW_GRAPH | KW_DIGRAPH) dot_id? '{' dot_stmt_list '}';
245
246 dot_stmt_list: ( dot_stmt ';' '?' ) *;
247
248 dot_stmt:
249     dot_node_stmt
250     | dot_edge_stmt
251     | dot_attr_stmt

```

```

252     | dot_id '=' dot_id
253     | dot_subgraph;
254
255 dot_attr_stmt: ( KW_GRAPH | KW_NODE | KW_EDGE ) dot_attr_list;
256
257 dot_attr_list: ( '[' dot_a_list? ']' )+;
258
259 dot_a_list: ( dot_id ( '=' dot_id )? ',' )? )+;
260
261 dot_edge_stmt: ( dot_node_id | dot_subgraph ) dot_edgeRHS dot_attr_list?;
262
263 dot_edgeRHS: ( dot_edgeop ( dot_node_id | dot_subgraph ) )+;
264
265 dot_edgeop: '->' | '--';
266
267 dot_node_stmt: dot_node_id dot_attr_list?;
268
269 dot_node_id: dot_id dot_port?;
270
271 dot_port: ':' dot_id ( ':' dot_id )?;
272
273 dot_subgraph: ( KW_SUBGRAPH dot_id )? '{' dot_stmt_list '}';
274
275 dot_id:
276     NAME
277     | STRING
278     | INT
279     | FLOAT
280     | KW_GRAPH
281     | KW_NODE
282     | KW_EDGE;
283
284 /* Lexer rules */
285 TYP_STRING: 'str';
286 TYP_INT: 'int';
287 TYP_FLOAT: 'float';
288 TYP_LIST: 'list';
289 TYP_DICT: 'dict';
290 TYP_BOOL: 'bool';
291 KW_TYPE: 'type';
292 KW_GRAPH: 'graph';
293 KW_STRICT: 'strict';
294 KW_DIGRAPH: 'digraph';
295 KW_SUBGRAPH: 'subgraph';
296 KW_NODE: 'node';
297 KW_IGNORE: 'ignore';
298 KW_TAKE: 'take';
299 KW_SPAWN: 'spawn';
300 KW_WITH: 'with';
301 KW_ENTRY: 'entry';
302 KW_EXIT: 'exit';
303 KW_LENGTH: 'length';
304 KW_KEYS: 'keys';
305 KW_CONTEXT: 'context';
306 KW_INFO: 'info';

```

```

307 KW_DETAILS: 'details';
308 KW_ACTIVITY: 'activity';
309 COLON: ':';
310 DBL_COLON: '::';
311 COLON_OUT: '::>';
312 LBRACE: '{';
313 RBRACE: '}';
314 KW_EDGE: 'edge';
315 KW_WALKER: 'walker';
316 SEMI: ';';
317 EQ: '=';
318 PEQ: '+=';
319 MEQ: '-=';
320 TEQ: '*=';
321 DEQ: '/=';
322 CPY_EQ: ':=';
323 KW_AND: 'and' | '&&';
324 KW_OR: 'or' | '||';
325 KW_IF: 'if';
326 KW_ELIF: 'elif';
327 KW_ELSE: 'else';
328 KW_FOR: 'for';
329 KW_TO: 'to';
330 KW_BY: 'by';
331 KW_WHILE: 'while';
332 KW_CONTINUE: 'continue';
333 KW_BREAK: 'break';
334 KW_DISENGAGE: 'disengage';
335 KW_SKIP: 'skip';
336 KW_REPORT: 'report';
337 KW_DESTROY: 'destroy';
338 DOT: '.';
339 NOT: '!' | 'not';
340 EE: '==';
341 LT: '<';
342 GT: '>';
343 LTE: '<=';
344 GTE: '>=';
345 NE: '!=';
346 KW_IN: 'in';
347 KW_ANCHOR: 'anchor';
348 KW_HAS: 'has';
349 KW_PRIVATE: 'private';
350 COMMA: ',';
351 KW_CAN: 'can';
352 PLUS: '+';
353 MINUS: '-';
354 MUL: '*';
355 DIV: '/';
356 MOD: '%';
357 POW: '^';
358 LPAREN: '(';
359 RPAREN: ')';
360 LSQUARE: '[';
361 RSQUARE: ']';

```

```
362 FLOAT: ([0-9]+)? '.' [0-9]+;  
363 STRING: '"' ~ ["\r\n"]* '"' | '\'' ~ ['\r\n']* '\'';  
364 BOOL: 'true' | 'false';  
365 INT: [0-9]+;  
366 NULL: 'null';  
367 NAME: [a-zA-Z_] [a-zA-Z0-9_]*;  
368 COMMENT: '/*' .*? '*/' -> skip;  
369 LINE_COMMENT: '//' ~[\r\n]* -> skip;  
370 PY_COMMENT: '#' ~[\r\n]* -> skip;  
371 WS: [\t\r\n] -> skip;  
372 ErrorChar: .;
```

Bibliography

- [1] Wikimedia Commons. File:baby in wikimedia foundation "hello world" onesie.jpg — wikimedia commons, the free media repository, 2020. [Online; accessed 29-July-2021].
- [2] Wikimedia Commons. File:directed graph no background.svg — wikimedia commons, the free media repository, 2020. [Online; accessed 13-July-2021].
- [3] Wikimedia Commons. File:multi-pseudograph.svg — wikimedia commons, the free media repository, 2020. [Online; accessed 9-July-2021].
- [4] Thomas H. Cormen, Charles E. Leiserson, Ronald L. Rivest, and Clifford Stein. *Introduction to Algorithms, 3rd Edition*. MIT Press, 2009.
- [5] Django Software Foundation. Django.
- [6] J. K. Rowling. *Harry Potter and the Philosopher's Stone*, volume 1. Bloomsbury Publishing, London, 1 edition, June 1997.
- [7] The Python Foundation. Pypi: The python package index.